

# D Templates: Some Notes

Philippe Sigaud

Wednesday 28<sup>th</sup> December, 2011

## Contents

<b>Introduction</b>	<b>6</b>
<b>What's in This Document</b>	<b>6</b>
<b>Conventions</b>	<b>6</b>
<b>How to Get This Document</b>	<b>7</b>
<b>I Basics</b>	<b>8</b>
<b>1 Template Declarations</b>	<b>8</b>
<b>2 Instantiating a Template</b>	<b>10</b>
<b>3 Templates Building Blocks</b>	<b>13</b>
3.1 The Eponymous Trick . . . . .	13
3.2 Inner <code>alias</code> . . . . .	14
3.3 <code>static if</code> . . . . .	15
3.3.1 Syntax . . . . .	15
3.3.2 Optional Code . . . . .	16
3.3.3 Nested <code>static ifs</code> . . . . .	16
3.3.4 Recursion with <code>static if</code> . . . . .	17
3.4 Templates Specializations . . . . .	20
3.5 Default Values . . . . .	22
<b>4 Function Templates</b>	<b>23</b>
4.1 Syntax . . . . .	23
4.2 <code>auto return</code> . . . . .	24
4.3 IFTI . . . . .	24
4.4 Example: Flattening Arrays and Ranges . . . . .	25
4.5 Anonymous Function Templates . . . . .	26
4.6 Functions Overloading . . . . .	28
4.7 Storage Classes . . . . .	28
4.8 Properties are Automatically Deduced . . . . .	29
4.9 <code>in</code> and <code>out</code> Clauses . . . . .	29

4.10	Modifying Functions: Memoizing a Function	29
4.11	Modifying Functions: Currying a Function	33
4.12	Modifying Functions: Expanding a function	34
<b>5</b>	<b>Struct Templates</b>	<b>34</b>
5.1	Syntax	34
5.2	Factory Functions	35
5.3	Giving Access to Inner Parameters	36
5.4	Templated Member Functions	36
5.5	Templated Constructors	39
5.6	Inner Structs	40
5.7	Template This Parameters	42
5.8	Example: a Concat / Flatten Range	42
<b>6</b>	<b>Class Templates</b>	<b>46</b>
6.1	Syntax	46
6.2	Methods Templates	47
6.3	<code>invariant</code> clauses	47
6.4	Inner Classes	48
6.5	Anonymous Classes	48
6.6	Parameterized Base Class	48
6.7	Another Example	49
6.8	The Curiously Recurring Template Pattern	51
6.9	Example	51
<b>7</b>	<b>Other Templates?</b>	<b>51</b>
7.1	Interface Templates	52
7.2	Union Templates	52
<b>II</b>	<b>Some More Advanced Considerations</b>	<b>53</b>
<b>8</b>	<b>Constraints</b>	<b>53</b>
8.1	Syntax	53
8.2	Constraints Usage	54
8.3	Constraints Limits	55
8.4	Constraints, Specializations and <code>static if</code> :	56
<b>9</b>	<b>Predicate Templates</b>	<b>57</b>
9.1	Testing for a member	57
9.2	Testing for operations	58
9.3	Completing the <code>Flatten</code> range:	58
<b>10</b>	<b>Template Tuple Parameters</b>	<b>59</b>
10.1	Definition and Basic Properties	59
10.2	The Type of Tuples	62
10.3	Example: Variadic Functions	63
10.4	One-Element Tuples: Accepting Types and Alias	65
10.5	Example: Inheritance Lists	65

<b>11 Operator Overloading</b>	<b>68</b>
11.1 Syntax	68
11.2 Example: Arithmetic Operators	68
11.3 Special Case: <code>in</code>	69
11.4 Special Case: <code>cast</code>	69
<b>12 Mixin Templates</b>	<b>69</b>
12.1 Syntax	69
12.2 Mixing Code In	70
<b>13 opDispatch</b>	<b>71</b>
13.1 Syntax	71
13.2 Getters and Setters	72
13.3 Wrapper Templates	73
<b>14 Templates All the Way Up</b>	<b>74</b>
<b>15 <code>__FILE__</code> and <code>__LINE__</code></b>	<b>74</b>
<b>III Around Templates: Other Compile-Time Tools</b>	<b>76</b>
<b>16 String Mixins</b>	<b>76</b>
16.1 Syntax	76
16.2 Mixing Code In, With Templates	76
16.3 Limitations	79
<b>17 Compile-Time Function Evaluation</b>	<b>79</b>
17.1 Evaluation at Compile-Time	79
17.2 <code>__ctfe</code>	80
17.3 Templates and CTFE	80
17.4 Templates and CTFE and String Mixins, oh my!	80
17.5 Simple String Interpolation	81
17.6 Example: extending <code>std.functional.binaryFun</code>	83
<b>18 <code>__traits</code></b>	<b>84</b>
18.1 Yes/No Questions with <code>__traits</code>	85
18.2 <code>__traits(identifier, XXX)</code>	85
18.3 <code>getMember</code>	86
18.4 <code>allMembers</code>	87
18.5 <code>derivedMembers</code>	90
18.6 <code>getOverloads</code>	90
18.7 Getting All Members, Even Overloaded Ones	90
18.8 Traits TODO	93
<b>19 Wrapping it all Together</b>	<b>93</b>
19.1 Cracking Open a Type	93
<b>IV Examples</b>	<b>94</b>

<b>20 Type Sorcery</b>	<b>94</b>
20.1 Mapping, Filtering and Folding Types . . . . .	94
20.1.1 Mapping on Type Tuples . . . . .	94
20.1.2 Example: Testing a Function . . . . .	95
20.1.3 Filtering Type Tuples . . . . .	96
20.1.4 Example: building a <b>Graph</b> . . . . .	97
20.1.5 Folding Type Tuples . . . . .	99
20.2 Zipping Types, Interspersing Types, Crossing Types . . . . .	99
<b>21 Relational Algebra</b>	<b>100</b>
<b>22 Cloning, sort of</b>	<b>100</b>
<b>23 Generic Maker Function</b>	<b>100</b>
<b>24 Recording Successive States</b>	<b>101</b>
<b>25 Fields</b>	<b>102</b>
<b>26 Extending an enum</b>	<b>103</b>
<b>27 Static Switching</b>	<b>104</b>
<b>28 Gobble</b>	<b>104</b>
<b>29 A Polymorphic Tree</b>	<b>104</b>
<b>30 Polymorphic Association Lists</b>	<b>104</b>
<b>31 Expression Templates</b>	<b>105</b>
<b>32 Wrapping a Function</b>	<b>105</b>
<b>33 Mapping n ranges in parallel</b>	<b>106</b>
<b>34 Statically-Checked Writeln</b>	<b>107</b>
<b>35 Extending a Class</b>	<b>107</b>
<b>36 Pattern Matching With Functions</b>	<b>107</b>
<b>37 Generating a Switch for Tuples</b>	<b>107</b>
<b>38 Tuples as Sequences</b>	<b>108</b>
38.1 Mapping on Tuples . . . . .	108
38.2 Filtering Tuples . . . . .	108
<b>Appendices</b>	<b>109</b>

<b>A</b>	<b>A Crash Course on the <code>is(...)</code> Expression</b>	<b>109</b>
A.1	General Syntax . . . . .	109
A.2	<code>is(Type)</code> . . . . .	109
A.3	<code>is(Type : AnotherType)</code> and <code>is(Type == AnotherType)</code> . . .	111
A.4	Type Specializations . . . . .	113
	<b>Index</b>	<b>115</b>

# Introduction

Templates are a central feature of D, giving you powerful compile-time code generation abilities that'll make your code cleaner, more flexible and even more efficient. They are used everywhere in [Phobos](#), D standard library and any D user should know about them. But, based on C++'s templates as they are, they can be a bit daunting at first. The [D Programming Language](#) website's [documentation](#) is a good start, though its description of templates is spread among many different files and (as it's a language reference) its material doesn't so much *teach* you how to use templates as *show* you their syntax and semantics.

This document aims to be a kind of tutorial on D templates, to show the beginning D coder what can be achieved with them. When I was doing C++, I remember *never* using templates for more than *containers-of-T* stuff, and considered Boost-level<sup>1</sup> metaprogramming the kind of code I could never understand, never mind produce. Well, D's sane syntax for templates, nifty things like `static if`, `alias` or tuples cured me of that impression. I hope this document will help you also.

## What's in This Document

**Part I** deals with the very basics: how to declare and instantiate a template, the standard 'building blocks' you'll use in almost all your templates, along with function (4), struct (5) and class (6) templates. Throughout the text, examples will present applications of these concepts.

**Part II** is about more advanced topics a D template user will probably use, but not on a daily basis, like template constraints (8), mixin templates (12) or operator overloading (11).

**Part III** presents other metaprogramming tools: string mixins (16), compile-time function evaluation (17) and `__traits` (18). These are seen from a template-y point of view: how they can interact with templates and what you can build with them in conjunction with templates.

**Part IV** presents more developed examples of what can be done with templates, based on real needs I had at some time and that could be fulfilled with templates.

Finally, an appendix ([Appendix A](#)) on the ubiquitous `is` expression completes this document.

## Conventions

In this doc:

- D keywords will be marked like this: `int`, `static if`, `__traits`.
- Symbols and names used in code samples and cited in the text will be written like this: `myFunc`, `flatten`.
- internal links will be in red, like this ([I](#)).

---

<sup>1</sup> The [Boost](#) C++ library collection makes heavy use of templates.

- external links will be in blue, like [this](#).

Code samples are shown in their own syntax-highlighted box:

```
1 import std.stdio;
2
3 // This is a comment
4 void hello(uint times)
5 {
6     foreach(i; 0..times) writeln("Hello, Word!");
7 }
```

For those of you interested by L<sup>A</sup>T<sub>E</sub>X, this document uses the [minted](#) package for the code samples. Yes, I know they are larger than the main text, but that's to get about 80-chars-long lines. And no, for the time being I'll just respect the standard L<sup>A</sup>T<sub>E</sub>X text width, unless many people complain about it.

I will sometimes make a little *aparte*, discussing a small piece of info too small to be in its own section. These will be marked so:

### What's with all that red and blue?

What, don't you like having red links and blue keywords in your text?

Finally, some sections in this doc are not finished yet. The sections I consider unfinished will begin with a friendly frame:

THIS SECTION IS STILL A WORK IN PROGRESS:  
That's also true for this very intro. I should have a 'Thanks' subsection where I list people that contributed to this text.

## How to Get This Document

This doc is just a bunch of L<sup>A</sup>T<sub>E</sub>X documents [hosted on GitHub](#). Don't hesitate to fork it or (even better for me), to make pull requests! For those of you reading this on paper, the adress is:

<http://github.com/PhilippeSigaud/D-templates-tutorial>

## Part I

# Basics

A template is a recipe, a blueprint that will generate code at your command and according to compile-time parameters you'll give. Templates are *parameterized code*. Each template definition is written once in a module and can then be instantiated many times with different parameters, possibly resulting in quite different code, depending on the arguments you used.

## 1 Template Declarations

Here is the syntax for a template declaration:

```
1 template templateName(list, of, parameters)
2 {
3     // Some syntactically correct declarations here
4     // The arguments are accessible inside the template scope.
5 }
```

`templateName` is your usual D identifier and the list of parameters is a comma-separated list of zero or more template parameters. These can be:

**Types:** An `identifier` alone by itself is considered a type name. The common D style is to use identifiers beginning with a capital letter (`Range`, `Rest`), as for any user-defined types. Many D templates use the C++ tradition of one-capital-letter names for types, starting from `T` (`U`, `V`, ...). Do not feel constrained by this, use what makes your templates most easy to understand.

**Aliases:** These will capture symbols: variable names, class names, even other template names. They will also accept many compile-time literals: strings, arrays, function literals, ... Mostly, if you need a widely-accepting template, use an alias parameter. They will *not* accept built-in types as arguments, however. You declare them with `alias identifier`.

**Literal values:** They can be integral values (`int`, `ulong`, ...), enum-based, strings, chars, floating-point values or boolean values. I think the rule is that any expression that can be evaluated at compile-time is OK. They are all declared like this: `typeName identifier`. For example: `int depth` or `string name`.

**Template parameters tuples:** Template parameters tuples will capture under one identifier an entire list of template parameters (types, names, literals, ...). Tuples will store any template argument you will throw at them. If no argument is passed, you will just get a zero-length tuple. Really, as they can deal with types as well as symbols, these tuples are a bit of a mongrel type but they are wonderfully powerful and easy to use, as you will see in section 10. The syntax is `identifier...` (yes, three dots) and tuples must be the last parameter of a template.



Of those, types and aliases are the most common, while floating point values are fairly rare (their use as arguments for compile-time calculations have been superseded by D's Compile-Time Function Evaluation, aka CTFE, described in section 17). You'll see different uses of these parameters in this document.

Note that pointers, arrays, objects (instantiated classes), structs or functions are not part of this list. But as I said, alias parameters allow you to capture and use array, class, function or struct *names* and then access their capacities.

### Aliases, Symbols and Names

There is big difference between built-in types (like `int` or `double[3]`) and user-defined types. A user-defined type, say a class called `MyClass`, is a type name. So, it's *both* a type (the class `MyClass`, accepted by type templates arguments) and a name (`MyClass`, accepted by `alias` template parameters). On the other hand, `int`, being a D keyword is not a symbol nor a name. It's just a type. You cannot pass it to an alias template parameter.

The template body can contain any standard D declarations: variable, function, class, interface, other templates, alias declarations, ... The only exception I can think of is declaring a `module`, as this is done at the top-level scope.

### Syntax and Semantics

And then there is a catch: code inside a `template` declaration must only by syntactically correct D code (that is: code that looks like D code). The semantics is checked only during instantiation. That means you can code happily, writing templates upon templates and the compiler won't bat an eye if you do not exercise your templates by instantiating them.

Inside the template body, the parameters are all accessible as placeholders for the future arguments. Also, the template's own name refers to its current instantiation when the code is generated. This is mostly used in struct (see section 5) and class (6) templates.

Here are some template declaration examples:

```
1 template ArrayOf(T) // T is a type
2 {
3     alias T[] ArrayType;
4     alias T ElementType;
5     immutable T t; // storage class
6
7 }
8
9 template Transformer(From, To) // From and To are types, too
10 {
11     To transform(From from) { /* some code that returns a To*/}
12
13     class Modifier
14     {
```

```

15         From f;
16         To t;
17         this(From f) { ... }
18     }
19 }
20
21 template nameOf(alias a)
22 {
23     enum string name = a.stringof; // enum: manifest constant
24                                     // determined at compile-time
25 }
26
27 template ComplicatedOne(T, string s, alias a, bool b, int i)
28 { /* some code using T, s, a, b and i */ }
29
30 template Minimalist() {} // Zero-parameter template declaration.
31
32 template OneOrMore(FirstType, Rest...) // Rest is a tuple.
33 { ... }
34
35 template ZeroOrMore(Types...) // Types is a tuple.
36 { ... }
37
38 template Multiple(T)      { ... } // One arg version.
39 template Multiple(T,U)    { ... } // Two args,
40 template Multiple(T,U,V)  { ... } // and three.

```

The real syntax for template declarations is slightly more complex, I'll introduce more of it in the next sections (You'll see for example type restrictions in section 3.4, default values in section 3.5, instantiation constraints in 8 and more on tuples in section 10).

There is a limitation that's interesting to keep in mind: templates can be declared in almost any scope, except inside a (standard) function.

### enum

In the previous code, see line 23? It defines a `string` called `name` as a member of `NameOf`. The `enum` placed right before means `name` is a compile-time constant. You can see it as a kind of storage class, in the line of `immutable` or `const`, one that means the value is totally defined and fixed at runtime. You'll see numerous examples of `enum` in this document.

## 2 Instantiating a Template

To instantiate a template, use the following syntax:

```

1 templateName!(list, of, arguments)

```

Note the ‘!’ before the comma-separated argument list. If the argument list contains only one argument (one token), you can drop the parenthesis:

```
1 templateName!argument
```

### Templates as templates arguments

Arguments can themselves be the result of another template instantiation. If a template returns a type upon instantiation, it’s perfectly OK to use it inside another template argument list. In this document you’ll regularly see Matrioshka calls like this: `firstTemp!(secondTempl!(Arguments), OtherArguments)`.

The compiler will have a look at the declarations (if more than one template were declared with the called name) and select the one with the correct number of arguments and the correct types to instantiate. If more than one template can be instantiated, it will complain and stop there (though, have a look on template specializations in section 3.4 and template constraints in section 8).

When you instantiate a template, the global effect is that a new named scope is created in the template declaration scope. The name of this new scope is the template name with its argument list: `templateName!(args)`. Inside the scope, the parameters are now ‘replaced’ with the corresponding arguments (storage classes get applied, variables are initialized, ...). Here’s what possible instantiations of the templates declared in section 1 might look like:

```
1 ArrayOf!int
2
3 Transformer!(double,int) // From is an alias for the type double
4                          // To for the type int
5
6 struct MyStruct { ... }
7 nameOf!(MyStruct) // "MyStruct" is a identifier -> captured by alias
8
9 ComplicatedOne!( int[]    // a type
10                  , "Hello" // a string literal
11                  , ArrayOf // a name (here the ArrayOf template)
12                  , true    // a boolean literal
13                  , 1+2      // calculated to be the integral '3'.
14                  )
15
16 Minimalist!()
17 // or even:
18 Minimalist
19
20 OneOrMore!( int           // FirstType is int.
21             , double, string, "abc" // Rest is (double,string,"abc")
22             )
23
24 ZeroOrMore!(int)          // Types is a 1-element tuple: (int)
```

```

25 ZeroOrMore!(int,double,string) // Types is (int,double,string)
26 ZeroOrMore!()                  // Types is the empty tuple: ()
27
28 Multiple!(int)                  // Selects the one-arg version
29 Multiple!(int,double,string) // The three args version.
30 Multiple!()                    // Error! No 0-arg version

```

Outside the scope (that is, where you put the template instantiation in your own code), the internal declarations are accessible by fully qualifying them:

```

1 // ArrayType is accessible (it's int[])
2 // array is a completely standard dynamic array of ints.
3 ArrayOf!(int).ArrayType array;
4 ArrayOf!(int).ElementType element; // the same, element is an int.
5
6 // the transform function is accessible. Instantiated like this,
7 // it's a function from string to double.
8 auto d = Transformer!(string,double).transform("abc"); // d is a double

```

Obviously, using templates like this, with their full name, is a pain. The nifty `D alias` declaration is your friend:

```

1 alias Transformer!(string,double) StoD;
2
3 auto d = StoD.transform("abc");
4 auto m = new StoD.Modifier("abc"); // StoD.Modifier is a class
5                                     // storing a string and a double.

```

You must keep in mind that instantiating a template means generating code. Using different arguments at different places in your code will instantiate *as many differently named scopes*. This is a major difference with *generics* in languages like Java or C#, where generic code is created only once and type erasure is used to link all this together. On the other hand, trying to instantiate many times the ‘same’ template (ie: with the same arguments) will only create one piece of code.

```

1 alias Transformer!(string,double) StoD;
2 alias Transformer!(double,string) DtoS;
3 alias Transformer!(string,int) StoI;
4 // Now we can use three different functions and three different classes.

```

### void

Note that `void` is a D type and, as such, a possible template argument for a type parameter. Just be careful because many templates make no sense when `void` is used as a type. In the following sections and in the appendix, you’ll see ways to restrict arguments to some types.

## 3 Templates Building Blocks

Up to now, templates must seem not that interesting to you, even with a simple declaration and instantiation syntax. But wait! D introduced a few nifty tricks that both simplify and greatly expand templates use. This section will introduce you to your future best friends, the foundations on which your templates will be built.

### 3.1 The Eponymous Trick

If a template declares only one symbol with the same name (greek: *epo-nymous*) as the enclosing template, that symbol is assumed to be referred to when the template is instantiated. This one is pretty good to clean your code:

```
1 template pair(T)
2 {
3     T[] pair(T t) { return [t,t];} // declares only pair
4 }
5
6 auto array = pair!(int)(1); // no need to do pair!(int).pair(1)
```

```
1 template NameOf(alias name)
2 {
3     enum string NameOf = name.stringof;
4 }
5
6 int foo(int i) { return i+1;}
7 auto s1 = NameOf!(foo); // s1 is "foo"
8 auto s2 = NameOf!(NameOf)
```

A limitation is that the eponymous trick works *only* if you define one (and only one) symbol. Even if the other symbols are private and such, they will break the eponymous substitution. This may change at a latter time, as the D developers have shown interest for changing this, but for the time being, you cannot do:

```
1 template ManyAlias(T, U)
2     T[] firstArray;
3     U[] secondArray;
4     T[U] ManyAlias; // Halas, ET substitution doesn't work here
5 }
6
7 // Hoping for ManyAlias!(int,string).ManyAlias
8 ManyAlias!(int, string) = ["abc":0, "def":1];
9
10 ManyAlias!(int, string).firstArray = [0,1,2,3];
```

But, will you ask, what if I want client code to be clean and readable by using the eponymous trick, when at the same time I need to internally create

many symbols in my template? In that case, create a secondary template with as many symbols as you need, and expose its final result through a (for example) `result` symbol. The first template can instantiate the second one, and refer only to the `.result` name:

```
1  /* your code */
2
3  // primary template
4  template MyTemplate(T, U, V)
5  {
6      // eponymous trick activated!
7      enum MyTemplate = MyTemplateImpl!(T,U,V).result;
8  }
9
10 // secondary (hidden) template
11 template MyTemplateImpl(T, U, V)
12 {
13     // The real work is done here
14     // Use as many symbols as you need.
15     alias symbol1 ...
16     enum otherName = ...
17     enum result = ...
18 }
```

```
1  /* client code */
2  MyTemplate!(int,string,double[]) someValue;
3  (...)
```

You can find an example of this two-templates idiom in Phobos, for example in `std.functional.unaryFun` or `std.functional.binaryFun`.

### 3.2 Inner alias

A common use for templates is to do some type magics: deducing types, assembling them in new way, etc. Types are not first-class entities in D (there is no ‘`type`’ type), but they can easily be manipulated as any other symbol, by aliasing them. So, when a template has to expose a type, it’s done by aliasing it to a new name.

```
1  template AllArraysOf(T)
2  {
3      alias T      Element;
4      alias T*     PointerTo;
5      alias T[]    DynamicArray;
6      alias T[1]   StaticArray;
7      alias T[T]   AssociativeArray;
8  }
```

## Exposing Template Parameters

Though they are part of a template's name, its parameters are *not* directly accessible externally. Keep in mind that a template name is just a scope name. Once it's instantiated, all the Ts and Us and such do not exist anymore. If you need them externally, expose them through a template member, as is done with `AllArraysOf.Element`. You will find other examples of this in section 5 on structs templates and section 6 on class templates.

### 3.3 `static if`

#### 3.3.1 Syntax

The `static if` construct<sup>2</sup> let you decide between two code paths at compile time. It's not specific to templates (you can use it in other part of your code), but it's incredibly useful to have your templates adapt themselves to the arguments. That way, using compile-time-calculated predicates based on the template arguments, you'll generate different code and customize the template to your need.

The syntax is:

```
1 static if (compileTimeExpression)
2 {
3     /* Code created if compileTimeExpression is evaluated to true */
4 }
5 else /* optional */
6 {
7     /* Code created if it's false */
8 }
```

Something really important here is a bit of compiler magic: once the code path is selected, the resulting code is instantiated in the template body, but without the curly braces. Otherwise that would create a local scope, hiding what's happening inside to affect the outside and would drastically limit the interest of `static if`. So the curly braces are only there to group the statements together.

If there is only one statement, you can get rid of the braces entirely, something you'll see frequently in D code. For example, suppose you need a template that 'returns' `true` if the passed type is a dynamic array and `false` otherwise (this kind of predicate template is developed a bit more in section 9).

```
1 template isDynamicArray(T)
2 {
3     static if (is(T t == U[], U))
4         enum isDynamicArray = true;
5     else
6         enum isDynamicArray = false;
7 }
```

<sup>2</sup> It's *both* an expression and a declaration, so I'll call it a construct.

As you can see, no curly braces after `static if` and we are also using the eponymous trick (`isArray` is the only symbol defined by the template and its type is automatically deduced by the compiler), resulting in a very clean syntax. The `is()` part is a way to get compile-time introspection which goes hand in hand with `static if`. There is a crash course on it at the end of this document (see [Appendix A](#)).

### 3.3.2 Optional Code

A common use of `static if` is to enable or disable code: a single `static if` without an `else` clause will generate code only when the condition is true. You can find many examples of this idiom in `std.range` where higher-level ranges (ranges wrapping other ranges) will activate some functionality only if the wrapped range can support it, like this:

```
1  /* We are inside a MyRange templated struct, wrapping an R. */
2
3      R innerRange;
4
5  /* some code that exist in all instantiations of MyRange */
6  (...)
7
8  /* optional code */
9  static if (hasLength!R) // does innerRange has a .length() method?
10     auto length()      // then MyRange has one also
11     {
12         return innerRange.length;
13     }
14
15 static if (isInfinite!R)    // Is innerRange an infinite range?
16     enum bool empty = false; // Then MyRange is also infinite.
17 // And so on...
```

### 3.3.3 Nested static ifs

`static ifs` can be nested: just put another `static if` after `else`. Here is a template selecting an alias:

```
1  import std.traits: isIntegral, isFloatingPoint;
2
3  template selector(T, alias intFoo, alias floatFoo, alias defaultFoo)
4  {
5      static if (isIntegral!T)
6          alias intFoo selector;
7      else static if (isFloatingPoint!T)
8          alias floatFoo selector;
9      else // default case
10         alias defaultFoo selector;
11 }
```

If you need a sort of `static switch` construct, see section [27](#).



### 3.3.4 Recursion with `static if`

**Rank:** Now, let's use `static if` for something a bit more complicated than just dispatching between code paths: recursion. What if you know you will receive n-dimensional arrays (simple arrays, arrays of arrays, arrays of arrays of arrays, ...), and want to use the fastest, super-optimized numerical function for the 1-dim array, another one for 2D arrays and yet another one for higher-level arrays. Abstracting this away, we need a template doing some introspection on types, that will return 0 for an element (anything that's not an array), 1 for a 1-dim array (`T[]`, for some `T`), 2 for a 2-dim array (`T[] []`), and so on. Mathematicians call this the *rank* of an array, so we will use that. The definition is perfectly recursive:

```
1 template rank(T)
2 {
3     static if (is(T t == U[], U)) // is T an array of U, for some type U?
4         enum rank = 1 + rank!(U); // then let's recurse down.
5     else // Base case, ending the recursion.
6         enum rank = 0;
7 }
```

Line 4 is the most interesting: with some `is` magic, `U` has been determined by the compiler and is accessible inside the `static if` branch. We use it to peel one level of `[]` off the type and recurse downward, using `U` as a new type for instantiating `rank`. Either `U` is itself an array (in which case the recursion will continue) or it will hit the base case and stop there.

```
1 static assert(rank!(int) == 0);
2 static assert(rank!(int[]) == 1);
3 static assert(rank!(int[] []) == 2);
4 static assert(rank!(int[] [] []) == 3);
5
6 /* It will work for any type, obviously */
7 struct S {}
8
9 static assert(rank!(S) == 0);
10 static assert(rank!(S[]) == 1);
11 static assert(rank!(S*) == 0);
```

#### `static assert`

Putting `static` before an `assert` forces the `assert` execution at compile-time. Using an `is` expression as the test clause gives assertion on types. One common use of `static assert` is to stop the compilation, for example if we ever get in a bad code path, by using `static assert(0, someString)`. The string is then emitted as a compiler error message.

**Rank for Ranges:** D has an interesting sequence concept that's called *range* and comes with predefined testing templates in `std.range`. Why not extend `rank`

to have it deal with ranges and see if something is a range of ranges or more? A type can be tested to be a range with `isInputRange` and its element type (the ‘U’) is obtained by applying `ElementType` to the range type. Both templates are found in `std.range`. Also, since arrays are included in the range concept, we can entirely ditch the array part and use only ranges. Here is a slightly modified version of `rank`:

```

1 import std.range;
2
3 template rank(T)
4 {
5     static if (isInputRange!T)                // is T a range?
6         enum rank = 1 + rank!(ElementType!T); // if yes, recurse
7
8     else                                        // base case, stop there
9         enum rank = 0;
10 }
11
12 auto c = cycle([[0,1],[2,3]]); // == [[0,1],[2,3],[0,1],[2,3],[0,1]...
13 assert(rank!(typeof(c)) == 2); // range of ranges

```

**Base Element Type:** With `rank`, we now have a way to get the number of `[]`’s in an array type (`T[] [] []`) or the level of nesting in a range of ranges. The complementary query would be to get the base element type, `T`, from any array of arrays ... of `T` or the equivalent for a range. Here it is:

```

1 template BaseElementType(T)
2 {
3     static if (rank!T == 0)                // not a range
4         static assert(0, T.stringof ~ " is not a range.");
5     else static if (rank!T == 1) // simple range
6         alias ElementType!T              BaseElementType;
7     else                                // at least range of ranges
8         alias BaseElementType!(ElementType!(T)) BaseElementType;
9 }

```

Line 4 is an example of `static assert` stopping compilation if we ever get into a bad code path. Line 8 is an example of a Matrioshka call: a template using another template’s call as its parameter.

**Generating Arrays:** Now, what about becoming more generative by inverting the process? Given a type `T` and a rank `r` (an `int`), we want to obtain `T[] [] ... []`, with `r` levels of `[]`’s. A rank of 0 means producing `T` as the result type.

```

1 template NDimArray(T, int r)
2 {
3     static if (r < 0)
4         static assert(0, "NDimArray error: the rank must be positive.");

```

```

5   else static if (r == 0)
6       alias T NDimArray;
7   else // r > 0
8       alias NDimArray!(T, r-1) [] NDimArray;
9   }

```

Here, recursion is done on line 8: we instantiate `NDimArray!(T,r-1)`, which is a type, then create an array of them by putting `[]` at the end and expose it through an alias. This is also a nice example of using an integral value as a template parameter.

```

1  alias NDimArray!(double, 8) Level8;
2  static assert(is(Level8 == double[] [] [] [] [] [] [] []));
3  static assert(is(NDimArray!(double, 0) == double));

```

**Repeated composition:** As a last example, we will use an alias template parameter in conjunction with some `static if` recursion to define a template that creates the ‘exponentiation’ of a function, its repeated composition. Here is what I mean by this:

```

1  string foo(string s) { return s ~ s;}
2
3  // power!(foo, n) is a function.
4  assert(power!(foo, 0)("a") == "a");           // identity function
5  assert(power!(foo, 1)("a") == foo("a"));       // "aa"
6  assert(power!(foo, 2)("a") == foo(foo("a"))); // "aaaa"
7  assert(power!(foo, 3)("a") == foo(foo(foo("a")))); // "aaaaaaaa"
8
9  // It's even better with function templates:
10 Arr[] makeArray(Arr)(Arr array) { return [array,array];}
11
12 assert(power!(makeArray, 0)(1) == 1);
13 assert(power!(makeArray, 1)(1) == [1,1]);
14 assert(power!(makeArray, 2)(1) == [[1,1],[1,1]]);
15 assert(power!(makeArray, 3)(1) == [[[1,1],[1,1]],[[1,1],[1,1]]]);

```

First, this is a template that ‘returns’ (becomes, rather) a function. It’s easy to do with the eponymous trick: just define inside the template a function with the same name. Secondly, it’s clearly recursive in its definition, with two base cases: if the exponent is zero, then we shall produce the identity function and if the exponent is one, we shall just return the input function itself. That being said, `power` writes itself:

```

1  template power(alias fun, uint exponent)
2  {
3      static if (exponent == 0) // degenerate case -> id function
4          auto power(Args)(Args args) { return args; }
5      else static if (exponent == 1) // end-of-recursion case -> fun

```

```

6     alias fun power;
7 else
8     auto power(Args...)(Args args)
9     {
10        return .power!(fun, exponent-1)(fun(args));
11    }
12 }

```

### **.power**

If you are wondering what's with the `.power` syntax on line 10, it's because by defining an eponymous template, we hide the parent template's name. So inside `power(Args...)` it refers to `power(Args...)` and not `power(alias fun, uint exponent)`. Here we want a new `power` to be generated so we call on the global `power` template with the 'global scope' operator (`.`).

In all three branches of `static if`, `power` exposes a `power` member, activating the eponymous template trick and allowing for an easy use by the client. Note that this template will work not only for unary (one argument) functions but also for n-args functions<sup>3</sup>, for delegates and for structs or classes that define the `()` (ie, `opCall`) operator and for function templates...<sup>4</sup>

Now, are you beginning to see the power of templates?

### **Curried Templates?**

No, I do not mean making them spicy, but separating the templates arguments, so as to call them in different places in your code. For `power`, that could mean doing `alias power!2 square;` somewhere and then using `square!fun1`, `square!fun2` at your leisure: the `exponent` parameter and the `fun` alias are separated. In fact, `power` is already partially curried: `fun` and `exponent` are separated from `Args`. For more on this, see section 14.

Given a template `temp`, writing a `curry` template that automatically generates the code for a curried version of `temp` is *also* possible, but outside the scope of this document.

## **3.4 Templates Specializations**

Up to now, when we write a `T` in a template parameter list, there is no constraint on the type that `T` can become during instantiation. Template specialization is a small 'subsyntax', restricting templates instantiations to a subset of all possible types and directing the compiler into instantiating a particular version of a

<sup>3</sup> Except for the degenerate,  $n = 0$  case, since the identity function defined above accepts only one arg. A more than one argument version is possible, but would need to return a tuple.

<sup>4</sup> I cheated a little bit there, because the resulting function accepts any number of arguments of any type, though the standard function parameters checks will stop anything untowards to happen. A cleaner (but longer, and for template functions, more complicated) implementation would propagate the initial function parameter tuple.

template instead of another. If you’ve read [Appendix A](#) on the `is` expression, you already know how to write them. If you didn’t, please do it now, as it’s really the same syntax. These specializations are a direct inheritance from C++ templates, up to the way they are written and they existed in D from the very beginning, long before `static if` or templates constraints were added.

The specializations are added in the template parameter list, the `(T, U, V)` part of the template definition. `Type : OtherType` restricts `Type` to be implicitly convertible into `OtherType` and `Type == OtherType` restricts `Type` to be exactly `OtherType`.

```

1 template ElementType(T == U[], U) // can only be instantiated with arrays
2 {
3     alias U ElementType;
4 }
5
6 template ElementType(T == U[n], U, size_t n) // only with static arrays
7 {
8     alias U ElementType;
9 }
10
11 // Say Array is a class, which has an ElementType type alias defined.
12 template ElementType(T : Array)
13 {
14     alias Array.ElementType ElementType;
15 }

```

Now, the `Type == AnotherType` syntax may seem strange to you: if you know you want to restrict `Type` to be *exactly* `AnotherType`, why make it a template parameter? It’s because of templates specializations’ main use: you can write different implementations of a template (with the same name, obviously) and when asked to instantiate one of them, the compiler will automatically decide which one to use based the ‘most adapted’ to the provided arguments. ‘Most adapted’ obeys some complicated rules you can find on the D Programming Language web site, but they act in a natural way most of the time. The neat thing is that you can define the most general template *and* some specialization. The specialized ones will be chosen when it’s possible. For

```

1 template InnerType(T : U*, U) // Specialization for pointers
2 {
3     alias U InnerType;
4 }
5
6 template InnerType(T : U[], U) // Specialization for dyn. arrays
7 { ... }
8
9 template InnerType(T) // Standard, default case
10 { ... }
11
12 int* p;
13 int i;

```

```

14 alias InnerType!(typeof(p)) Pointer; // pointer spec. selected
15 alias InnerType!(typeof(i)) Default; // standard template selected

```

This idiom is used frequently in C++, where there is no (built-in) `static if` construct or template constraints. Oldish D templates used it a lot, too, but since other ways have been around for some years, recent D code seems to be more constraint-oriented: have a look at heavily templated Phobos modules, for example `std.algorithm` or `std.range`.

### Specializations or `static if` or Templates Constraints?

Yes indeed. Let's defer this discussion for when we have seen all three sub-systems.

## 3.5 Default Values

Like functions parameters, templates parameters can have default values. The syntax is the same: `Param = defaultValue`. The default can be anything that makes sense with respect to the parameter kind: a type, a literal value, a symbol or another template parameter.

```

1 template Default(T = int, bool flag = false)
2 { ... }
3
4 Default!(double);           // Instantiate Default!(double, false)
5 Default!(double, true);     // Instantiate Default!(double, true) (Doh!)
6 Default!();                 // Instantiate Default!(int, false)

```

A difference with function parameters default values is that, due to specializations (3.4) or IFTI (4.3), some parameters can be automatically deduced by the compiler. So, default templates parameters are not forced to be the last parameters:

```

1 template Deduced(T : U[], V = U, U)
2 { ... }
3
4 Deduced!(int[], double); // U deduced to be int. Force V to be a double.
5 Deduced!(int[]);        // U deduced to be int. V is int, too.
6 Deduced!();              // Error, T is not some array.

```

### Specialization and Default Value?

Yes you can. Put the specialization first, then the default value. Like this: `(T : U[] = int[], U)`. It's not commonly used, though.

As for functions, well-chosen defaults can greatly simplify standard calls. See for example the `std.algorithm.sort`. It's parameterized on a predicate and a swapping strategy, but both are adapted to what most people need when sorting. That way, most client uses of the template will be short and clean, but customization to their own need is still possible.

**TODO:** Maybe something on template dummy parameters, like those used by `std.traits.ReturnType`. Things like `dummy == void`.

## 4 Function Templates

### 4.1 Syntax

If you come from languages with generics, maybe you thought D templates were all about parameterized classes and functions and didn't see any interest in the previous sections (acting on types?). Fear not, you can also do type-generic functions and such in D, with the added generative power of templates.

As we have seen in section 3.1, if you define a function inside a template and use the template's own name, you can call it easily:

```
1 // declaration:
2 template myFunc(T, int n)
3 {
4     auto myFunc(T t) { return to!int(t) * n;}
5 }
6
7 // call:
8 auto result = myFunc!(double,3)(3.1415);
9
10 assert(result == to!int(3.1415)*3);
```

Well, the true story is even better. First, D has a simple way to declare a function template: just put a template parameter list before the argument list:

```
1 string concatenate(A,B)(A a, B b)
2 {
3     return to!string(a) ~ to!string(b);
4 }
5
6 Arg select(string how = "max", Arg)(Arg arg0, Arg arg1)
7 {
8     static if (how == "max")
9         return (arg0 < arg1) ? arg1 : arg0;
10    else static if (how == "min")
11        return (arg0 < arg1) ? arg0 : arg1;
12    else
13        static assert(0,
14            "select: string 'how' must be either \"max\" or \"min\".");
15 }
```

Nice and clean, uh? Notice how the return type can be templated too, using `Arg` as a return type in `select`.

## 4.2 `auto` return

Since you can select among code paths, the function return type can vary widely, depending on the template parameters you passed it. Use `auto` to simplify your code:<sup>5</sup>

```
1 auto morph(T, U, alias f)(U arg)
2 {
3     static if (is(U == int) && is(T == class))
4     {
5         return new T(f(arg));
6     }
7     else static if (is(T == function))
8     {} // void-returning function
9     else static if (...)
10    (...)
11 }
```

### `auto ref`

A function template can have an `auto ref` return type. That means that for templates where the returned values are rvalues, the template will get the `refed` version. And the non-`ref` version if not.

## 4.3 IFTI

Even better is Implicit Function Template Instantiation (IFTI), which means that the compiler will most of the time be able to automatically determine a template parameters by studying the function arguments. If some template arguments are pure compile-time parameters, just provide them directly:

```
1 /* suppose the previous concatenate(A,B) function */
2 string res1 = concatenate(1, 3.14); // A is int and B is double
3
4 struct Foo {}
5 string res2 = concatenate("abc", Foo()); // A is string, B is Foo
6
7 /* suppose the previous select(string how = "max", Arg) function */
8 auto res3 = select(3,4); // how is "max", Arg is int.
9 auto res4 = select!"min"(3.1416, 2.718); // how is "min", Arg is double.
```

As you can see, this results in very simple calling code. So we can both declare function templates and call them with a very clean syntax. The same

---

<sup>5</sup> `auto` return functions used to have some limitations, like for example not appearing in docs. I remember having to code type-manipulating templates to get correct the return type.



can be done with structs or classes and such, as you will see in the next sections. In fact, the syntax is so clean that, if you are like me, you may forget from time to time that you are *not* manipulating a function (or a struct, etc.): you are manipulating a template, a parameterized piece of code.

### A Mantra

XXX templates are not XXXs, they are templates. With XXX being any of (function, struct, class, interface, union). Templates are parameterized scopes and scopes are not first-class in D: they have no type, they cannot be assigned to a variable, they cannot be returned from functions. That means, for example, that you *cannot* return function templates, you cannot inherit from class templates and so on. Of course, *instantiated* templates are perfect examples of functions, classes, and such. Then you can inherit, return...

We may encounter The Mantra again in this tutorial.

## 4.4 Example: Flattening Arrays and Ranges

Let's use what we have just seen in a concrete way. In D, you can manipulate 2D, 3D arrays, but sometimes need to process them linearly. As of this writing, neither `std.algorithm` nor `std.range` provide a `flatten` function. Beginning with simple arrays, here is what we want:

```
1 assert( flatten([[0,1],[2,3],[4]]) == [0,1,2,3,4] );
2 assert( flatten([[0,1]]) == [0,1] );
3 assert( flatten([0,1]) == [0,1] );
4 assert( flatten(0) == 0 );
5
6 assert( flatten([[[0,1],[2]], [[3], [4,5]], [[]], [[6,7,8]]])
7           == [0,1,2,3,4,5,6,7,8] );
8
9 assert( flatten([[[[[[[]]]]]]]) == [] ); // void[] [] [] [] [] -> void[]
```

So, studying the examples, we want a non-array or a simple array to be unaffected by `flatten`: it just returns them. For arrays of rank 2 or higher, it collapses the elements down to a rank-1 array. It's classically recursive: we will apply `flatten` on all subarrays with `std.algorithm.map` and concatenate the elements with `std.algorithm.reduce`:

```
1 import std.algorithm;
2
3 auto flatten(Arr)(Arr array)
4 {
5     static if (rank!Arr == 0)
6         return [array];
7     else
8     {
9         auto children = map!(&flatten)(array);
```

```

10     return reduce!"a~b"(children); // concatenate the children
11 }
12 }

```

We make good use of D `auto` return parameter for functions there. In fact, a single call to `flatten` will create one instance per level, all with a different return type.

Note that `flatten` works perfectly on ranges too, but is not lazy: it eagerly concatenates all the elements down to the very last one in the innermost range. Ranges being lazy, a good `flatten` implementation for them should itself be a range that delivers the elements one by one, calculating the next one only when asked to (and thus, would work on infinite or very long ranges too, which the previous simple implementation cannot do). Implementing this means creating a struct template (5) with a factory function (5.2). You will find this as an example in section 5.8.

From our current `flatten`, it's an interesting exercise to add another parameter: the number of levels you want to flatten. Only the first three levels or last two innermost, for example. Just add an integral template parameter that gets incremented (or decremented) when you recurse and is another stopping case for the recursion. Positive levels could mean the outermost levels, while a negative argument would act on the innermost ones. A possible use would look like this:

```

1 flatten!1([[0,1],[]], [[2]], [[3], [4,5]], [[]], [[6,7,8]]);
2     // == [[0,1],[],[2],[3], [4,5],[],[6,7,8]]
3 flatten!2([[0,1],[]], [[2]], [[3], [4,5]], [[]], [[6,7,8]]);
4     // == [0,1,2,3,4,5,6,7,8]
5 flatten!0([[0,1],[]], [[2]], [[3], [4,5]], [[]], [[6,7,8]]);
6     // ==[[0,1],[], [[2]], [[3], [4,5]], [[]], [[6,7,8]]]
7 flatten!(-1)([[0,1],[]], [[2]], [[3], [4,5]], [[]], [[6,7,8]]);
8     // ==[[0,1]], [[2]], [[3,4,5]], [[]], [[6,7,8]]]

```

## 4.5 Anonymous Function Templates

In D, you can define anonymous functions (delegates even, that is: closures):

```

1 auto adder(int a)
2 {
3     return (int b) { return a+b;};
4 }
5
6 auto add1 = adder(1); // add1 is an int delegate(int)
7 assert(add1(2) == 3);

```

In the previous code, `adder` returns an anonymous delegate. Could `Adder` be templated? Ha! Remember The Mantra (page 25): function templates are templates and cannot be returned. For this particular problem, there are two possible solutions. Either you do not need any new type and just use T:

```

1 auto adder(T)(T a)
2 {
3     return (T b) { return a+b;};
4 }
5
6 auto add1f = adder(1.0) // add1f is an float delegate(float)
7 assert(add1f(2.0) == 3.0);
8
9 import std.bigint;
10
11 // addBigOne accepts a BigInt and returns a BigInt
12 auto addBigOne = adder(BigInt("1000000000000000"));
13 assert(addBigOne(BigInt("1")) == BigInt("1000000000000001"));
14
15 // But:
16 auto error = add1(3.14); // Error! Waiting for an int, getting a double.

```

In the previous example, the returned anonymous delegate is *not* templated. It just happens to use T, which is perfectly defined once instantiation is done. If you really need to return something that can be called with any type, use an inner struct (see section 5.6).

Now, it may come as a surprise to you that D *does* have anonymous function templates. The syntax is a purified version of anonymous functions:

```

1 (a,b) { return a+b; }

```

Yes, the previous skeleton of a function is an anonymous template. But, remember The Mantra: you cannot return them. And due to (to my eyes) a bug in `alias` grammar, you cannot alias them to a symbol:

```

1 alias (a){ return a; } Id; // Error

```

So what good are they? You can use them with template alias parameters, when these stand for functions and function templates:

```

1 template callTwice(alias fun)
2 {
3     auto callTwice(T)(T t)
4     {
5         return fun(fun(t));
6     }
7 }
8
9 alias callTwice!( (a){ return a+1; }) addTwo;
10
11 assert(addTwo(2) == 4);

```

Since they are delegates, they can capture local symbols:

```

1 enum b = 3; // Manifest constant, initialized to 3
2
3 alias callTwice!( (a){ return a+b;}) addTwoB;
4
5 assert(addTwoB(2) == 2 + 3 + 3);

```

## 4.6 Functions Overloading

THIS SECTION IS STILL A WORK IN PROGRESS:  
OK, I need to write something on this.

## 4.7 Storage Classes

As seen in section 2, storage classes get applied to types during instantiation. It also works for function templates arguments:

```

1 void init(T)(ref T t)
2 {
3     t = T.init;
4 }
5
6 int i = 10;
7 init(i);
8 assert(i == 0);

```

Should the need arises, that also means you can customize your storage classes according to templates arguments. There is no built-in syntax for that, so you'll have to resort to our good friend `static if` and the eponymous trick:

```

1 // Has anyone a better example?
2 template init(T)
3 {
4     static if (is(T == immutable) || is(T == const))
5         void init(T t) {} // do nothing
6     else static if (is(T == class))
7         void init(ref T t)
8         {
9             t = new T();
10        }
11    else
12        void init(ref T t)
13        {
14            t = T.init;
15        }
16 }

```

## 4.8 Properties are Automatically Deduced

In D, a function can have the following properties:

- A function can be tagged with the `pure` property, which means it does not have side-effects: the value you get back is the only thing that matters.
- They can also be tagged with `@safe`, `@trusted` and `@system`. `@safe` means a function cannot corrupt memory. A `@trusted` function can call `@safe` ones, but offers no other guaranty concerning memory. And a `@system` function just laugh at you.
- The last property is `nothrow` which means you guarantee the function does not throw any exception.

As the compiler gets complete access to a function template code, it can analyze it and automatically deduce properties for you. This feature is still quite new as of this writing, but it does seem it works. So, *all* of your function templates will get a smattering of properties when they are instantiated (these properties will of course vary with the template parameters).

## 4.9 `in` and `out` Clauses

The `in` and `out` clauses for a function are given full access to templates parameters. As for other parameterized code, that means you can use `static if` to enable or disable code, depending on the template arguments.

```
1 import std.complex, std.math, std.traits;
2
3 auto squareRoot(N n) if (isNumeric!N || isComplex!N)
4 in
5 {
6     static if (isNumeric!N)
7         enforce(n > 0);
8     // no need to do that for a complex.
9 }
10 body
11 {
12     return sqrt(n);
13 }
```

## 4.10 Modifying Functions: Memoizing a Function

THIS SECTION IS STILL A WORK IN PROGRESS:  
Almost there, though.

TODO: Should this section be moved in the struct templates section?

Let us use a template to wrap a function and provide some additional usefulness when calling the function. Memoizing is an interesting and useful example:

if the function does long calculations, it might be efficient to store the computed results in an external structure and to query this structure for the result instead of calling the function again.

We have not seen struct templates yet (they are presented in section 5), but the following example should be easy to understand: the previous result are stored in an associative array, indexed on tuples of arguments. To get a function return type or parameter type tuple, just use Phobos' `std.traits.ReturnType` and `std.traits.ParameterTypeTuple`, which are templates that accept function *names* or types.

```

1 struct Memoize(alias fun)
2 {
3     alias ReturnType!fun RT;
4     alias ParameterTypeTuple!fun PTT;
5     RT[Tuple!(PTT)] memo; // stores the result, indexed by arguments.
6
7     RT opCall(PTT args) {
8         if (tuple(args) in memo) { // Have we already seen these args?
9             return memo[tuple(args)]; // if yes, use the stored result
10        }
11        else { // if not, compute the result and store it.
12            RT result = fun(args);
13            memo[tuple(args)] == result;
14            return result;
15        }
16    }
17 }
18
19 Memoize!fun memoize(alias fun)()
20 {
21     return Memoize!fun();
22 }

```

Usage is very simple:

```

1 int veryLongCalc(int i double d, string s) { ... }
2
3 auto vlcMemo = memoize!(veryLongCalc);
4
5 // calculate veryLongCalc(1, 3.14, "abc")
6 // takes minutes!
7 int res1 = vlcMemo(1, 3.14, "abc");
8 int res2 = vlcMemo(2, 2.718, "def"); // minutes again!
9 int res3 = vlcMemo(1, 3.14, "abc"); // a few ms to get res3

```

The above code is trivial and could be optimized in many ways. Mostly, a real memoizing template should also modify its behavior with storing policies. For example:

- No-limit or limited size store?

- In case of limited-size store: how to define the limit and what should be the eviction policy?
  - First-in/First-out memo?
  - Least recently used memo?
  - Least used?
  - Time-to-live?
  - Discard all and flush the store?
  - Discard only a fraction?
  - Stop memoizing?

The last X results could be stored in a queue: each time a result is pushed into the associative array, push the arguments tuples in the queue. Once you reach the maximum store limit, discard the oldest one or (for example) half the stored values.

Here is a possible small implementation. It makes for a nice example of enabling/disabling code with `static if` and `enum`-based policies. Note that I use D dynamic arrays as a primitive queue. A real queue could probably be more efficient, but there isn't one in the standard library as of this writing.

```

1 enum MemoStoringPolicy {
2     always, // there is no tomorrow
3     maximum // sustainable growth
4 }
5
6 enum MemoDiscardingPolicy {
7     oldest, // only discard the oldest result
8     fraction, // discard a fraction (0.5 == 50%)
9     all // burn, burn!
10 }
11
12 struct Memoize(alias fun,
13                 MemoStoringPolicy storing,
14                 MemoDiscardingPolicy discarding)
15 {
16     alias ReturnType!fun RT;
17     alias ParameterTypeTuple!fun PTT;
18
19     static if (storing == MemoStoringPolicy.maximum)
20     {
21         Tuple!(PTT)[] argsQueue;
22         size_t maxNumStored;
23     }
24
25     static if (discarding == MemoDiscardingPolicy.fraction)
26         float fraction;
27
28     RT[Tuple!(PTT)] memo; // stores the result, indexed by arguments.
29

```

```

30 RT opCall(PTT args) {
31     if (tuple(args) in memo) { // Have we already seen these args?
32         return memo[tuple(args)]; // if yes, use the stored result
33     }
34     else { // if not,
35         RT result = fun(args); // compute the result and store it.
36         memo[tuple(args)] == result;
37         return result;
38     }
39 }
40 }

```

And a few factory function to help creating those Memoize structs:

```

1 // No runtime arg -> always store
2 Memoize!(fun, MemoStoringPolicy.always, MemoDiscardingPolicy.all)
3 memoize(alias fun)()
4 {
5     return Memoize!(fun,
6         MemoStoringPolicy.always,
7         MemoDiscardingPolicy.all)();
8 }
9
10 // One runtime size_t arg -> maximum store / discarding all
11 Memoize!(fun, MemoStoringPolicy.maximum, MemoDiscardingPolicy.all)
12 memoize(alias fun)(size_t max)
13 {
14     return Memoize!(fun,
15         MemoStoringPolicy.maximum,
16         MemoDiscardingPolicy.all)(max);
17 }
18
19 // Two runtime args (size_t, double) -> maximum store / discarding a fraction
20 Memoize!(fun, MemoStoringPolicy.maximum, MemoDiscardingPolicy.fraction)
21 memoize(alias fun)(size_t max, double fraction)
22 {
23     return Memoize!(fun,
24         MemoStoringPolicy.maximum,
25         MemoDiscardingPolicy.fraction)(max, fraction);
26 }
27
28 // One compile-time argument (discarding oldest), one runtime argument (max)
29 Memoize!(fun, MemoStoringPolicy.maximum, discarding)
30 memoize(alias fun, MemoDiscardingPolicy discarding == MemoDiscardingPolicy.oldest)
31 (size_t max)
32 {
33     return Memoize!(fun,
34         MemoStoringPolicy.maximum,
35         discarding)(max);

```



36 }

Most of the time, the type of runtime arguments is enough to determine what you want as a memoizing/storing behavior. Only for the (rarer?) policy of discarding only the oldest stored result does the user need to indicate it with a template argument:

```
1 int veryLongCalc(int i double d, string s) { ... }
2
3 // Store the first million results, flush the memo on max
4 auto vlcMemo1 = memoize!(veryLongCalc)(1_000_000);
5
6 // Store the first million results, flush half the memo on max
7 auto vlcMemo2 = memoize!(veryLongCalc)(1_000_000, 0.5f);
8
9 // Store first twenty results, discard only the oldest
10 auto vlcMemo3 = memoize!(veryLongCalc, MemoDiscardingPolicy.oldest)(20);
```

## 4.11 Modifying Functions: Currying a Function

THIS SECTION IS STILL A WORK IN PROGRESS:  
Some explanations would greatly help there.

```
1 template CheckCompatibility(T...)
2 {
3     template With(U...)
4     {
5         static if (U.length != T.length)
6             enum With = false;
7         else static if (T.length == 0) // U.length == 0 also
8             enum With = true;
9         else static if (!is(U[0] : T[0]))
10            enum With = false;
11        else
12            enum With = CheckCompatibility!(T[1..$]).With!(U[1..$]);
13    }
14 }
```

```
1 struct Curry(alias fun, int index = 0)
2 {
3     alias ReturnType!fun RT;
4     alias ParameterTypeTuple!fun PTT;
5     PTT args;
6
7     auto opCall(V...)(V values)
8         if (V.length > 0
```

```

9      && V.length + index <= PTT.length)
10  {
11      // Is fun directly callable with the provided arguments?
12      static if (__traits(compiles, fun(args[0..index], values)))
13          return fun(args[0..index], values);
14      // If not, the new args will be stored. We check their types.
15      else static if (!CheckCompatibility!(PTT[index..index + V.length]).With!(V))
16          static assert(0, "curry: bad arguments. Waited for "
17                        ~ PTT[index..index + V.length].stringof
18                        ~ " but got " ~ V.stringof);
19      // not enough args yet. We store them.
20      else
21      {
22          Curry!(fun, index+V.length) c;
23          foreach(i,a; args[0..index]) c.args[i] = a;
24          foreach(i,v; values) c.args[index+i] = v;
25          return c;
26      }
27  }
28 }
29
30 auto curry(alias fun)()
31 {
32     Curry!(fun,0) c;
33     return c;
34 }

```

## 4.12 Modifying Functions: Expanding a function

THIS SECTION IS STILL A WORK IN PROGRESS:

I want to put some small function-wrapping templates: making a function that accepts more args than the original, or accepting arguments of a different type (that's useful when mapping on tuples, like in section 38).

## 5 Struct Templates

### 5.1 Syntax

As you might have guessed, declaring a struct template is done like this:

```

1 struct Tree(T)
2 {
3     T value;
4     Tree[] children;
5
6     bool isLeaf() @property { return children.empty;}

```

```

7  /* More tree functions: adding children, removing some,... */
8  }

```

### Tree[] or Tree!(T) []?

Remember that inside a template declaration, the template's name refers to the current instantiation. So inside `Tree(T)`, the name `Tree` refers to a `Tree!T`.

This gives us a run-of-the-mill generic tree, which is created like any other template:

```

1  auto t0 = Tree!int(0);
2  auto t1 = Tree!int(1, [t0,t0]);
3  Tree!int[] children = t1.children;

```

As with all previous templates, you can parameterize your structs using much more than simple types:

```

1  bool lessThan(T)(T a, T b) { return a<b;}
2
3  struct Heap(Type, alias predicate = lessThan, float reshuffle = 0.5f)
4  {
5      // predicate governs the internal comparison
6      // reshuffle deals with internal re-organizing of the heap
7      Type[] values;
8      (...)
9  }

```

Struct templates are heavily used in `std.algorithm` and `std.range` for lazy iteration, have a look there.

## 5.2 Factory Functions

Now, there is one limitation: struct constructors do not do activate IFTI (4.3) like template functions do. In the previous subsection to instantiate `Tree(T)` I had to explicitly indicate `T`:

```

1  auto t0 = Tree!int(0); // Yes.
2
3  auto t1 = Tree(0); // Error, no automatic deduction that T is int.

```

This is because templated constructors are possible (see 5.5) and may have template parameters differing from that of the global struct template. But honestly that's a pain, even more so for struct templates with many template arguments. There is a solution, of course: use a template function to create the correct struct and return it. Here is an example of such a factory function for `Tree`:

```

1 auto tree(T)(T value, T[] children = null)
2 {
3     return Tree!(T)(value, children);
4 }
5
6 auto t0 = tree(0); // Yes!
7 auto t1 = tree(1, [t0,t0]); // Yes!
8
9 static assert(is( typeof(t1) == Tree!int ));
10
11 auto t2 = tree(t0); // Yes! typeof(t2) == Tree!(Tree!(int))

```

Once more, have a look at `std.algorithm` and `std.range`, they show numerous examples of this idiom.

### 5.3 Giving Access to Inner Parameters

As was said in section 3.2, template arguments are not accessible externally once the template is instantiated. For the `Tree` example, you might want to get an easy access to `T`. As for any other templates, you can expose the parameters by aliasing them. Let's complete our `Tree` definition:

```

1 struct Tree(T)
2 {
3     alias T Type;
4     T value;
5     Tree[] children;
6
7     bool isLeaf() @property { return children.empty;}
8 }
9
10 auto t0 = tree("abc");
11 alias typeof(t0) T0;
12
13 static assert(is( T0.Type == string ));

```

### 5.4 Templated Member Functions

A struct template is a template like any other: you can declare templates inside, even functions templates. Which means you can have templated member functions.

**Mapping on a Tree** Let us use that to give our `Tree` a mapping ability. For a range, you can use `std.algorithm.map` to apply a function in turn to each element, thus delivering a transformed range. The same process can be done for a tree, thereby keeping the overall *shape* but modifying the elements.<sup>6</sup>

<sup>6</sup> We could easily make that a free function, but this is the member function section.

Let's think a little bit about it before coding. `map` should be a function template that accepts any function name as a template alias parameter (like `std.algorithm.map`). Let's call this alias `fun`. The `value` member should be transformed by `fun`, that's easy to do. We want to return a new `Tree`, which will have for type parameter the result type of `fun`. If `fun` transforms `As` into `Bs`, then a `Tree!A` will be mapped to a `Tree!B`. However, since `fun` can be a function template, it may not have a pre-defined return type that could be obtained by `std.traits.ReturnType`. We will just apply it on a `T` value (obtained by `T.init` and take this type. So `B` will be `typeof(fun(T.init))`.

What about the children? We will map `fun` on them too and collect the result into a new children array. They will have the same type: `Tree! (B)`. If the mapped `Tree` is a leaf (ie: if it has no children), the process will stop.

Since this is a recursive template, we have to help the compiler a bit with the return type. Here we go:<sup>7</sup>

```
1  /* rest of Tree code */
2      Tree!(typeof(fun(T.init))) map(alias fun)()
3      {
4          alias typeof(fun(T.init)) MappedType;
5          MappedType mappedValue = fun(value);
6          Tree!(MappedType)[] mappedChildren;
7          foreach(child; children) mappedChildren ~= child.map!(fun);
8          return tree(mappedValue, mappedChildren);
9      }
```

Let's use it:

```
1  auto t0 = tree(0);
2  auto t1 = tree(1, [t0,t0]);
3  auto t2 = tree(2, [t1, t0, tree(3)]);
4
5  /* t2 is
6            2
7          / | \
8         1 0 3
9        / \
10     0  0
11 */
12 // t2 is a Tree!(int)
13 static assert(is( t2.Type == int ));
14
15 // Adding one to all values
16 int addOne(int a) { return a+1;}
17 auto t3 = t2.map!(add(1));
18
19 /* t3 is
20           3
21         / | \
22        2 1 4
```

<sup>7</sup> The difference with Phobos `map` is that it's not lazy.

```

22         / \
23        1  1
24    */
25
26    assert(t3.value = 3);
27
28    // Converting all values to strings
29    import std.conv:to;
30    auto ts = t2.map!(to!string); // we convert every value into a string;
31
32    /* ts is          "2"
33                   / | \
34                  "1" "0" "3"
35                   / \
36                  "0"  "0"
37    */
38
39    assert(is( ts.Type == string ));
40    assert(ts.value == "2");

```

**Folding a Tree** You may feel `map` is not really a member function: it does not take any argument. Let's make another transformation on **Trees**: folding them, that is: collapsing all values into a new one. The range equivalent is `std.algorithm.reduce` which collapses an entire (linear) range into one value, be it a numerical value, another range or what have you

For a tree, folding can for example generate all values in pre-order or post-order, calculate the height of the tree, the number of leaves... As for ranges, folding is an extremely versatile function. It can in fact be used to convert a **Tree** into an array or another **Tree**. We will do just that.

Taking inspiration from `reduce`, we need an seed value and *two* folding functions. The first one, `ifLeaf`, will be called on childless nodes, for `fold` to return `ifLeaf(value, seed)`. The second one, `ifBranch`, will be called nodes with children. In this case, we first apply `fold` on all children and then return `ifBranch(value, foldedChildren)`. In some simple cases, we can use the same function, hence a default case for `ifBranch`. Here is the code:<sup>8</sup>

```

1  /* rest of Tree code */
2  auto fold(alias ifLeaf, alias ifBranch = ifLeaf, S)(S seed)
3  {
4      if (isLeaf)
5      {
6          return ifLeaf(value, seed);
7      }
8      else
9      {
10         typeof(Tree.init.fold!(ifLeaf, ifBranch)(seed))[] foldedChildren;
11         foreach(child; children)

```

<sup>8</sup> Technically, `std.algorithm.reduce` is a *left fold*, while what is shown here is a *right fold*. The difference is not essential to this document.

```

12         foldedChildren ~= child.fold!(ifLeaf, ifBranch)(seed);
13         return ifBranch(value, foldedChildren);
14     }
15 }

```

Let's play a bit with it. First, we want to sum all values of a tree. For leaves, we just return the node's `value` plus `seed`. For branches, we are given `value` and an array containing the sums of values for all children. We need to sum the values of this array, add it to the node's `value` and return that. In that case, we do not care for the seed.

```

1 auto sumLeaf(T, S)(T value, S seed)
2 {
3     return value + seed;
4 }
5
6 auto sumBranch(T)(T value, T[] summedChildren)
7 {
8     return value + reduce!"a+b"(summedChildren);
9 }
10
11 auto sum = t2.fold!(sumLeaf, sumBranch)(0);
12 assert(sum == 2 + (1 + 0 + 0) + 0 + 3);

```

In the same family, but a bit more interesting is getting all values for in-order iteration: given a tree node, return an array containing the local value and then the values for all nodes, recursively.

```

1 T[] inOrderL(T, S)(T value, S seed)
2 {
3     return [value] ~ seed;
4 }
5
6 T[] inOrderB(T)(T value, T[] [] inOrderChildren)
7 {
8     return [value] ~ reduce!"a~b"(inOrderChildren);
9 }
10
11 int[] seed; // empty array
12 auto inOrder = t2.fold!(inOrderL, inOrderB)(seed);
13 assert(inOrder == [2, 1, 0, 0, 0, 1, 3]);

```

And as a last use, why not build a tree?

**TODO: Write just that.**

## 5.5 Templated Constructors

Struct constructors are member function, so they can be templated too. They need not have the same template parameters than their mother struct:

```

1 struct S(T)
2 {
3     this(U)(U u) { ... }
4 }
5
6 auto s = S!string(1); // T is string, U is int.

```

As you can see, IFTI (4.3) works for constructors. `U` is automatically deduced, though you have to indicate `T` in this case. The previous example is drastically limited: you cannot have any value of type `U` in the struct, because `U` does not exist outside the constructor. A bit more useful would be to collect an alias (a function, for example) and use it to initialize the struct. If it's used only for initialization, it can be discarded afterwards. But then, IFTI is not activated by an alias...

The most interesting use I've seen is to make conversions during the struct's construction:

```

1 struct Holder(Type)
2 {
3     Type value;
4
5     this(AnotherType)(AnotherType _value)
6     {
7         value = to!Type(_value);
8     }
9 }
10
11 Holder!int h = Holder!int(3.14);
12 assert(h.value == 3);

```

That way, a `Holder!int` can be constructed with any value, but if the conversion is possible, it will always hold an `int`.

## 5.6 Inner Structs

You can create and return inner structs and use the local template parameters in their definition. We could have a factory function for `Heap` like this:

```

1 auto heap(alias predicate, Type)(Type[] values)
2 {
3     struct Heap
4     {
5         Type[] values;
6         this(Type[] _values)
7         {
8             /* some code initializing values using predicate */
9         }
10        /* more heapy code */
11    }

```



```

12 |
13 |     return Heap(values); // alias predicate is implicit there
14 | }

```

In that case, the `Heap` struct is encapsulated inside `heap` and uses the `predicate` alias inside its own engine, but it's not a templated struct itself. I did not use `Tree` as an example, because with recursive types it becomes tricky.

By the way, strangely enough, though you cannot declare 'pure' templates inside functions, you can declare struct templates. Remember the `adder` function in section 4.5? It didn't need to be templated with one type for each argument, as most of the time when you add numbers, they have more or less the same type. But what about a function that converts its arguments to `strings` before concatenating them?

```

1 | auto concatenate(A)(A a)
2 | {
3 |     /* !! Not legal D code !!
4 |     return (B)(B b) { return to!string(a) ~ to!string(b); }
5 | }

```

The previous example is not legal D code. Of course, there is a solution: just return a struct with a template member function (5.4), in that case the `opCall` operator:

```

1 | auto concatenate(A)(A a)
2 | {
3 |     struct Concatenator
4 |     {
5 |         A a;
6 |
7 |         auto opCall(B)(B b)
8 |         {
9 |             return to!string(a) ~ to!string(b);
10 |        }
11 |    }
12 |
13 |    Concatenator c;
14 |    c.a = a; // So as not to activate opCall()
15 |
16 |    return c;
17 | }
18 |
19 | auto c = concatenate(3.14);
20 | auto cc = c("abc");
21 | assert(cc == "3.14abc");

```

See section 11 for more on operators overloading.

What about templated inner structs inside struct templates? It's perfectly legal:

```

1 struct Outer(O)
2 {
3     O o;
4
5     struct Inner(I)
6     {
7         O o;
8         I i;
9     }
10
11     auto inner(I)(I i) { return Inner!(I)(o,i);}
12 }
13
14 auto outer(O)(O o) { return Outer!(O)(o);}
15
16 auto o = outer(1); // o is an Outer!int;
17 auto i = o.inner("abc"); // Outer.Outer!(int).Inner.Inner!(string)
18 }

```

## 5.7 Template This Parameters

THIS SECTION IS STILL A WORK IN PROGRESS:  
Some kind of example would be great here.

Inside a struct or class template, there is another kind of template parameter: the template **this** parameter, declared with **this identifier**. **identifier** then gets the type of the **this** reference. It's useful mainly for two uses:

- mixin templates (12), where you do not know the enclosing type beforehand. Please see this section for some examples.
- to determine how the type of the **this** reference is qualified (**const**, **immutable**, **shared**, **inout** or unqualified).

## 5.8 Example: a Concat / Flatten Range

We will use what we've seen for a struct template a lazy range that flattens ranges of ranges into linear ranges. Remember the **flatten** function from section 4.4? It worked quite well but was *eager*, not *lazy*: given an infinite range (a cycle, for example) it would choke on it. We will here make a lazy flattener.

If you look at the ranges defined in `std.range`, you will see that most (if not all) of them are structs. That's the basic way to get laziness in D: the struct holds the iteration state and exposes the basic range primitives. At the very least to be an *input range* — the simplest kind of range —, a type must have the following members (be they properties, member functions or manifest constants):

**front:** Which returns the range's first element.

**popFront:** Which discards the first element and advances the range by one step.

**empty:** Which returns `true` if the range has no more element, `false` otherwise.

From this simple basis, powerful algorithms can be designed that act on ranges. D defines more refined range concepts by adding other constraints. A *forward range* adds the `save` member that's used to store a range internal state and allows an algorithm to start again from a saved position. A *bidirectional range* also has the `back` and `popBack` primitives for accessing the end of the range, and so on.

Here we will begin by creating a simple input range that takes a range of ranges and iterate on the inner elements. Let's begin with the very basis:

```
1 import std.range;
2
3 struct Flatten(Range)
4 {
5     Range range;
6     ...
7 }
8
9 auto flatten(Range)(Range range)
10 {
11     static if (rank!Range == 0)
12         static assert(0, "flatten needs a range.");
13     else static if (rank!Range == 1)
14         return range;
15     else
16         return Flatten!(Range)(range);
17 }
```

So we have a struct template and its associated factory function. It doesn't make sense to instantiate `Flatten` with any old type, so `Range` is checked to be a range, using the `rank` template we saw on page 17. We haven't seen template constraints yet (they are described in section 8), but they would be a good fit there too.

A range of ranges can be represented like this:

```
1 [ subRange1[elem11, elem12,...]
2   , subRange2[elem21, elem22,...]
3   , ... ]
```

We want `Flatten` to return elements in this order: `elem11, elem12, ...elem21, elem22, ...`. Note that for ranges of rank higher than 2, the `elemxys` are themselves ranges. At any given time, `Flatten` is working on a sub-range, iterating on its elements and discarding it when it's empty. The iteration will stop when the last subrange has been consumed, that is when `range` itself is empty.

```

1 struct Flatten(Range)
2 {
3     alias ElementType!Range    SubRange;
4     alias ElementType!SubRange Element;
5
6     Range range;
7     SubRange subRange;
8
9     this(Range _range) {
10         this.range = range;
11         discardEmptySubRanges();
12     },
13
14     Element front() { return subRange.front;}
15
16     bool empty() { return range.empty;}
17
18     void popFront() {
19         if (!subRange.empty) subRange.popFront;
20         discardEmptySubRanges();
21     }
22
23     void discardEmptySubRanges() {
24         while(subRange.empty && !range.empty) {
25             range.popFront;
26             if (!range.empty) subRange = range.front;
27         }
28     }
29 }

```

- I cheat a little bit with D standard bracing style, because it eats vertical space like there is no tomorrow.
- We begin on line 3 and 4 by defining some new types used by the methods. They are not strictly necessary but make the code easier to understand and expose these types to the outer world, if they are needed.
- A constructor is now necessary to correctly initialize the struct.
- `front` returns a subrange element.
- The `discardEmptySubRanges` function does what it says on the can: we do not iterate on empty subranges.

Let see if this template works, by creating an infinite range and giving it to `flatten`:

```

1 import std.range;
2
3 auto cy = cycle(["Hello", "World"]); // "Hello","World","Hello","World",...

```

```

4 auto flattened = flatten(cy);
5 assert(flattened.front == 'H');
6
7 auto takeTwelve = take(flattened, 12);
8 assert(array(takeTwelve) == "HelloWorldHe");

```

But then, this only works for ranges of ranges (of rank  $\leq 2$ ). We want something that flattens ranges of any rank down to a linear range. This is easily done, we just add recursion in the factory function:

```

1 auto flatten(Range)(Range range)
2 {
3     static if (rank!Range == 0)
4         static assert(0, "flatten needs a range.");
5     else static if (rank!Range == 1)
6         return range;
7     else static if (rank!Range == 2)
8         return Flatten!(Range)(range);
9     else // rank 3 or higher
10        return flatten(Flatten!(Range)(range));
11 }

```

And, testing:

```

1 auto rank3 = [[[0,1,2],[3,4,5],[6]]
2               ,[[7],[],[8,9],[10,11]]
3               ,[[],[ ]]
4               ,[[12]] ];
5
6 auto flat = flatten(rank3);
7 assert(rank!(typeof(flat)) == 1); // Yup, it's a linear range
8 assert(equal( flat, [0,1,2,3,4,5,6,7,8,9,10,11,12] ));
9
10 auto reallyFlat = flatten(flat);
11 assert(equal( reallyFlat, flat )); // No need to insist
12
13 import std.string, std.algorithm;
14
15 auto text =
16 "Sing, O goddess, the anger of Achilles son of Peleus,
17 that brought countless ills upon the Achaeans.
18 Many a brave soul did it send hurrying down to Hades,
19 and many a hero did it yield a prey to dogs and vultures,
20 for so were the counsels of Jove fulfilled
21 from the day on which the son of Atreus, king of men,
22 and great Achilles, first fell out with one another.";
23 auto lines = text.splitLines; // array of strings
24 string[][] words;
25 foreach(line; lines) words ~= array(splitter(line, ' '));

```

```

26 assert( rank!(typeof(words)) == 3); // range of range of strings
27                                     // range of range of array of chars
28 auto flat = flatten(words);
29
30 assert(equal(take(flat, 50),
31             "Sing,Odyssey,theangerofAchillesonofPeleus,thatbr"));

```

Here it is. It works and we used a struct template (this section, 5), `static if` (3.3), inner member alias (3.2), factory functions (5.2) and IFTI (4.3).

## 6 Class Templates

### This Section Needs You!

I'm not an OOP programmer and am not used to create interesting hierarchies. If anyone reading this has an interesting example of class templates that could be used throughout the section, I'm game.

### 6.1 Syntax

No surprise there, just put the template parameters list between `class` and the optional inheritance indication.

```

1 class MyClass(Type, alias fun, bool b = false)
2   : Base, Interface1, Interface2
3 { ... }

```

What's more fun is that you can have parameterized inheritance: the various template parameters are defined before the base class list, you can use them here:

```

1 class MyClass(Type, alias fun, bool b = false)
2   : Base!(Type), Interface1, Interface2!(fun,b)
3 { ... }

```

### Interface Templates?

Yes you can. See section 7

This opens interesting vistas, where what a class inherits is determined by its template arguments (since `Base` may be many different classes or even interfaces depending on `Type`). In fact, look at this:

```

1 enum WhatBase { Object, Interface, BaseClass }
2
3 template Base(WhatBase whatBase = WhatBase.Object)
4 {

```

```

5 static if (is(T == WhatBase.Object))
6     alias Object Base; // MyClass inherits directly from Object
7 else static if(is(T == WhatBase.Interface))
8     alias TheInterface Base;
9 else
10     alias TheBase Base;
11 }

```

With this, `MyClass` can inherit either from `Object`, D root to the class hierarchy, or from an interface or from another class. Obviously, the dispatching template could be much more refined. With a second template parameter, the base class could itself be parameterized, and so on.

What this syntax *cannot* do however is change the number of interfaces at compile-time.<sup>9</sup> It's complicated to say: 'with *this* argument, `MyClass` will inherit from I, J and K and with *that* argument, it will inherit only from L.' You'd need the previous interfaces to all participate in the action, to all be templates and such. If the needed interfaces are all pre-defined and not templated, you need wrapping templates. It's a pain. However, type tuples can be used to greatly simplify this (see section 10.5 for an example).

## 6.2 Methods Templates

An object's methods are nothing more than delegates with a reference to the local `this` context. As seen for structs (5.4), methods can be templates too.

THIS SECTION IS STILL A WORK IN PROGRESS:  
I need to write something on overriding methods with templates. Also, I need to find some interesting method example. Man, I do not do classes.

## 6.3 invariant clauses

In the same family than `in/out` clauses for functions (section 4.9), a class template's `invariant` clause has access to the template parameter. You cannot make it disappear totally, but you can get it to be empty with a `static if` statement.

```

1 class MyClass(T, U, V)
2 {
3     (...)
4
5     invariant
6     {
7         static if (/* some condition on T,U,V */ )
8         {
9             /* invariant code */
10        }

```

<sup>9</sup> Except, maybe, by having an interface template be empty for certain parameters, thus in effect disappearing from the list.

```

11     else
12     { /* empty invariant */ }
13 }
14 }

```

## 6.4 Inner Classes

It's the same principle than for structs (5.6). You can define inner classes using the template parameters. You can even give them methods templates that use other template arguments. There is really nothing different from inner structs.

## 6.5 Anonymous Classes

In D, you can return anonymous classes directly from a function or a method. Can these be templated? Well, they cannot be class templates, that wouldn't make sense. But you can return anonymous classes with templated methods, if you really need to.

```

1  // stores a function and a default return value.
2  auto acceptor(alias fun, D)(D default)
3  {
4      return new class
5      {
6          auto opCall(T)(T t)
7          {
8              static if (__traits(compiles, fun(T.init)))
9                  return fun(t);
10             else
11                 return default;
12         }
13     };
14 }
15
16 int add1(int i) { return i+1;}
17 auto accept = acceptor!(add1)(0);
18 auto test1 = accept(1);
19 assert(test1 == 2);
20
21 auto test2 = accept("abc");
22 assert(test2 == 0); // default value

```

**TODO: Test this!**

For `__traits(compiles, ...)`, see [here](#).

## 6.6 Parameterized Base Class

You can use a template parameter directly as a base class:

```

1 interface ISerializable
2 {

```



```

3     size_t serialize() @property;
4 }
5
6 class Serializable(T) : T, ISerializable
7 {
8     size_t serialize() @property { ... }
9 }

```

In this example, a `Serializable!SomeClass` can act as a `SomeClass`. It's not different from what you would do with normal classes except the idiom is now abstracted on the base class: you write the template once, it can then be used on any class.

If you have different interfaces like this, you can nest these properties:

```

1 auto wrapped = new Serializable!(Iterable!(SomeClass))(...);

```

Of course, the base class and the interface may themselves be parameterized:

```

1 enum SerializationPolicy { policy1, policy2 }
2
3 interface ISerializable
4 (SerializationPolicy policy = SerializationPolicy.policy1)
5 {
6     static if (is(policy == SerializationPolicy.policy1))
7         ...
8     else
9         .../
10 }
11
12 class Serializable(T, Policy) : T, ISerializable!Policy
13 {
14     ...
15 }

```

In D, you can also get this kind of effect with an `alias X this;` declaration in your class or struct. You should also have a look at mixin templates (12) and wrapper templates (13.3) for other idioms built around the same need.

## 6.7 Another Example

THIS SECTION IS STILL A WORK IN PROGRESS:  
I need to see with Timon what I can write there.

```

1 /**
2  Timon Gehr timon.gehr@gmx.ch via puremagic.com
3  This is an useful pattern.
4  I don't have a very useful example at hand, but this one should do.

```

```

5 | It does similar things that can be achieved with traits in Scala for example.
6 | */
7 | import std.stdio;
8 | abstract class Cell(T){
9 |     abstract void set(T value);
10 |    abstract const(T) get();
11 | private:
12 |     T field;
13 | }
14 |
15 | class AddSetter(C: Cell!T,T): C{
16 |     override void set(T value){field = value;}
17 | }
18 | class AddGetter(C: Cell!T,T): C{
19 |     override const(T) get(){return field;}
20 | }
21 |
22 | class DoubleCell(C: Cell!T,T): C{
23 |     override void set(T value){super.set(2*value);}
24 | }
25 |
26 | class OneUpCell(C: Cell!T,T): C{
27 |     override void set(T value){super.set(value+1);}
28 | }
29 |
30 | class SetterLogger(C:Cell!T,T): C{
31 |     override void set(T value){
32 |         super.set(value);
33 |         writeln("cell has been set to '",value,"'!");
34 |     }
35 | }
36 |
37 | class GetterLogger(C:Cell!T,T): C{
38 |     override const(T) get(){
39 |         auto value = super.get();
40 |         writeln("'",value,"' has been retrieved!");
41 |         return value;
42 |     }
43 | }
44 |
45 | class ConcreteCell(T): AddGetter!(AddSetter!(Cell!T)){}
46 | class OneUpDoubleSetter(T): OneUpCell!(DoubleCell!(AddSetter!(Cell!T))){}
47 | class DoubleOneUpSetter(T): DoubleCell!(OneUpCell!(AddSetter!(Cell!T))){}
48 | void main(){
49 |     Cell!string x;
50 |     x = new ConcreteCell!string;
51 |     x.set("hello");
52 |     writeln(x.get());
53 |
54 |     Cell!int y;

```

```

55     y = new SetterLogger!(ConcreteCell!int);
56     y.set(123); // prints: "cell has been set to '123'!"
57
58     y = new GetterLogger!(DoubleCell!(ConcreteCell!int));
59     y.set(1234);
60     y.get(); // prints "'2468' has been retrieved!"
61
62     y = new AddGetter!(OneUpDoubleSetter!int);
63     y.set(100);
64     writeln(y.get()); // prints "202"
65
66     y = new AddGetter!(DoubleOneUpSetter!int);
67     y.set(100);
68     writeln(y.get()); // prints "201"
69
70     // ...
71 }

```

## 6.8 The Curiously Recurring Template Pattern

```

1 class Base(Child) { ... }
2
3 class Derived : Base!Derived { ... }

```

Hold on, what does that mean? `Base` is easy to understand. But what about `Derived`? It says it inherits from another class that is templated on... `Derived` itself? But `Derived` is not defined at this stage! Or is it? Yes, that works. It's called CRTP, which stands for Curiously Recurring Template Pattern (see [Wikipedia](#) on this). But what could be the interest of such a trick?

As you can see in the [Wikipedia](#) document it's used either to obtain a sort of compile-time binding or to inject code in your derived class. For the latter, D offers mixin templates (12) which you should have a look at. CRTP comes from C++ where you have multiple inheritance. In D, I fear it's not so interesting. Feel free to prove me wrong, I'll gladly change this section.

## 6.9 Example

THIS SECTION IS STILL A WORK IN PROGRESS:  
 I want to add an example with `duplicator`, a template that takes a class and creates another class which is its clone: same base, same interfaces, etc.

## 7 Other Templates?

Two other aggregate types in D can be templated using the same syntax: interfaces and unions.

## 7.1 Interface Templates

The syntax is exactly what you might imagine:

```
1 interface Interf(T)
2 {
3     T foo(T);
4     T[] bar(T,int);
5 }
```

Templated interfaces are sometimes useful but as they look very much like class templates, I won't describe them. As before, remember The Mantra (see page 25): interface templates are *not* interfaces.

## 7.2 Union Templates

Here is the syntax, no surprise there:

```
1 union Union(A,B,C) { A a; B b; C c;}
```

Union templates seem like a good idea, but honestly I've never seen one. Any reader of this document, please give me an example if you know one.

Strangely, enumerations do not have the previous simplified syntax. To declare a templated enumeration, use the eponymous template trick:

```
1 template Enum(T)
2 {
3     enum Enum : T { A, B, C}
4 }
```

## Part II

# Some More Advanced Considerations

In the previous part, we saw what everyone should know about D templates. But in fact, there is much more to them than that. What follows is not necessarily more complicated, but it's probably a little less commonly used. As this document matures, some subjects may flow from [Part I](#) into [Part II](#) and the other way round.

## 8 Constraints

Templates constraints are a way to block a template instantiation if some condition is not met. Any condition that can be determined at compile-time is authorized, which makes constraints a superset of templates specializations (see [3.4](#)). As such, their usage grew rapidly once they were introduced and, if Phobos is any indication, templates specializations are on the contrary becoming less common.

### 8.1 Syntax

To obtain a constraint, put an `if` clause just after the template parameter list, and before the enclosed scope:

```
1 template templateName(T,U,V) if (someCondition on T, U or V)
2 {
3     ...
4 }
```

When the compiler tries to instantiate a template, it will first check the constraint. If it evaluates to `false`, the template declaration is not part of the considered set. That way, using constraints, you can keep or drop templates at your leisure. `is` expressions are your friend there, allowing you to get compile-time introspection on types. See the appendix ([A](#)) for a crash course on it.

You may have many template declarations with the same name and differing constraints (in fact, that's the very use case for constraints). Depending on the activated constraints, some or all will be considered by the compiler.

```
1 template Constrained(T)
2     if (is(T : int)) { ... } // #1
3 template Constrained(T)
4     if (is(T : float)) { ... } // #2
5 template Constrained(T,U)
6     if (is(T : int) && !is(U : float)) { ... } // #3
7 template Constrained(T,U)
8     if (is(T : int) && is(U : float)) { ... } // #4
9
```

```

10 Constrained!(byte) // #1
11 Constrained!(string) // Error, no declaration fits (string)
12 Constrained!(int,string) // #3 and #4 considered, but #4 is dropped.
13                          // So #3 it is.

```

This syntax is the same for the special-cases templates seen in sections 4, 5, 6 and 7. The only tricky part is for class templates, where you may wonder where to put the constraint: before or after the inheritance list? The answer is: before.

```

1 T theFunction(T)(T argument)
2   if (is(T : int) || is(T : double)) { ... }
3
4 struct TheStruct(T)
5   if (is(T : int) || is(T : double)) { ... }
6
7 class TheClass(T)
8   if (is(T : int) || is(T : double))
9   : BaseClass!T, Interface1 { ... }

```

When you write constraints, just remember they are a compile-time construct. For `theFunction`, `argument` is not known at compile-time, only its type, `T`. So you should not use `argument` in your constraint. If you need a value of type `T`, use `T.init`. For

```

1 auto callTwice(alias fun, T)(T arg)
2   // Is it OK to do fun(fun(some T))?
3   if (is(typeof({ fun(fun(T.init)); }()))))
4   {
5     return fun(fun(arg));
6   }

```

## 8.2 Constraints Usage

Constraints come from the same idea than C++0x `concept`, er..., `concept`, although simpler to define, understand and, as shown by D, implement. The idea is to define a set of conditions a type must respect to be a representative of a ‘concept’, and check for it before instantiating.

Have a look at constraints poster-child: *ranges*.<sup>10</sup> They were rapidly described in section 4.4.

`std.range` defines a set of templates that check the different ranges concepts, called `isInputRange`, `isForwardRange`... I call these `bool`-becoming templates *predicate templates* and talk about them in section 9. Usage is quite simple:

```

1 import std.range;
2
3 struct RangeWrapper(Range)

```

<sup>10</sup> Ranges are overdue a tutorial.

```

4 // Does Range comply with the input range 'concept'?
5 if (isInputRange!Range)
6 {
7     /* Here we know that Range has at least three member functions:
8        .front(), .popFront() and .empty(). We can use them happily.*/
9 }
10
11 // In the factory function too.
12 auto rangeWrapper(Range) if (isInputRange!Range)
13 {
14     return RangeWrapper!(Range)( ... );
15 }

```

In fact, it's a bit like a sort of compile-time interface or compile-time duck-typing: we do *not* care about `Range`'s 'kind': it may be a `struct` or a `class` for all we know. What is important is that it respects the *input range* concept.

The good news is that the compiler will complain<sup>11</sup> when it cannot instantiate a template due to constraints being not respected. It gives better error messages this way (although not as good as you might need).

### 8.3 Constraints Limits

The main problem is that, compared to templates specializations, you cannot do:

```

1 template Temp(T) if (is(T:int)) // #1
2 { ... } // specialized for ints
3
4 template Temp(T) // #2
5 { ... } // generic case
6
7 Temp!int // Error!

```

Why an error? Because the compiler finds that both the `int`-specialized and the generic version can be instantiated. It cannot decide which one you mean and, quite correctly, does nothing, humble code that it is. No problem, says you, we will just add a constraint to the generic version:

```

1 template Temp(T) if (is(T:int)) // #1
2 { ... } // specialized for ints
3
4 template Temp(T) if (!is(T:int)) // #2
5 { ... } // generic case
6
7 Temp!int // Works!

```

Now, when you try to instantiate with an `int`, template #2 is not present (its constraint is false and was dropped from the considered template list) and

<sup>11</sup> I think *compiler* and *complain* must have the same root.

we can have #1. Hurrah? Not quite. The #1-constraint wormed its way into the generic version, adding code where none was initially. Imagine you had not one, but three different specialized versions:

```

1 template Temp(T) if (is(T:int[])) // #1a
2 { ... } // specialized for arrays of ints
3
4 template Temp(T) if (isRange!T) // #1b
5 { ... } // specialized for ranges
6
7 template Temp(T) if (is(T:double[n], int n)) // #1c
8 { ... } // specialized for static arrays of double
9
10 template Temp(T) // #2, generic
11     if ( /* What constraint? */
12 { ... }

```

OK, quick: what constraint for #2? The complement to *all* other constraints. See:

```

1 template Temp(T) // #2, generic
2     if ( !(is(T:int[]))
3         && !isRange!T
4         && !is(T:double[n], int n))
5 { ... }

```

That's becoming complicated to maintain. If someone else adds a fourth specialization, you need to add a fourth inverted version of its constraint. Too bad, you still compile and calls `Temp: Temp!(int[])`. And there: error! Why? Because constraints #1a and #1b are not mutually exclusive: an `int[]` is also an input range. Which means that for #1b, you need to add a clause excluding arrays of `int` and maybe modify constraint #2.

Ouch.

So, yes, constraints are wonderful, but they do have drawbacks. As a data point, this author uses them all the time, even though specializations (3.4) are sometimes more user-friendly: most of what I want to impose and check on my types cannot be done by specializations.

## 8.4 Constraints, Specializations and `static if`:

I mean, come on! Three different ways to decide if your template exists or not?

```

1 template Specialized(T : U[], U)
2 { ... }
3
4 template Constrained(T) if (is(T : U[], U))
5 { ... }
6
7 template StaticIfied(T)
8 {

```



```

9   static if (is(T : U[], U))
10  { ... }
11  else // stop compilation
12      static assert(0, "StaticIfied cannot be instantiated.");
13 }

```

What were the D designers thinking? Well, they got specializations (3.4) from D's cousin, C++. The two other subsystems were added a few years later, as the power of D compile-time metaprogramming became apparent and more powerful tools were needed. So, the 'modern' subsystems are constraints and `static if` (3.3). Constraints are much more powerful than specializations, as anything you can test with specialization, you can test with an `is` expression in a constraint. And `static if` is wonderfully useful outside of template instantiation, so these two are well implanted in D and are there to stay. What about specializations, now? First, they are quite nice to have when porting some C++ code. Second, they have a nice effect that constraints do *not* have: when more than one definition could be instantiated, priority is given to the more specialized. You saw the explanation in the previous subsection.

So in the end, the conclusion is a bit of *D Zen*: you are given tools, powerful tools. As these are powerful, they sometimes can do what other options in your toolbox can do also. D does not constrain (!) you, chose wisely.

## 9 Predicate Templates

When you find yourself typing again and again the same `is` expression or the same complicated constraint, it's time to abstract it into another template, a `bool`-becoming one. If you have a look a section A.2, you'll see a way to test if a particular piece of D code is OK (compilable) or not. Another way to obtain this is by using `__traits(compiles, some Code)`.

In Part IV, the section 20.1.3, on page 98 shows another example of a predicate template.

### 9.1 Testing for a member

For example, if you want to test is some type can be serialized, through a `size_t serialize()` member function:

```

1  template isSerialazable(Type)
2  {
3      static if (__traits(compiles, {
4          Type type;
5          size_t num = type.serialize;
6          }))
7          enum bool isSerialazable = true;
8      else
9          enum bool isSerialazable = false;
10 }

```

## 9.2 Testing for operations

As seen in previous sections (5.8, 8.2), we are writing here a kind of compile-time interface. Any type can pass this test, as long as it has a `.serialize` member that returns a `size_t`. Of course, you're not limited to testing member functions. Here is a template that verify some type has arithmetic operations:

```
1 template hasArithmeticOperations(Type)
2 {
3     static if (__traits(compiles,
4         {
5             Type t;
6             t + t; // addition
7             t / t; // subtraction
8             t * t; // multiplication
9             t / t; // division
10            +t;    // unary +
11            -t;    // unary -
12        })
13     enum bool hasArithmeticOperations = true;
14 else
15     enum bool hasArithmeticOperations = false;
16 }
17
18 static assert(hasArithmeticOperation!int);
19 static assert(hasArithmeticOperation!double);
20
21 struct S {}
22 static assert(!hasArithmeticOperation!S);
```

As you can see, you can test for any type of D code, which means it's *much* more powerful than templates specializations (3.4) or the `is` expression (Appendix A).

You may also get a certain feel of a... *pattern* emerging from the previous two examples. All the scaffolding, the boilerplate, is the same. And we could easily template it on what operator to test, for example. It's possible to do that, but it means crafting code at compile-time. Wait until you see string mixins (16) and CTFE (17) in Part III.

## 9.3 Completing the Flatten range:

Let's come back to `Flatten` from section 5.8. Using concept-checking templates, we will verify the range-ness of the wrapper type and promote `Flatten` to forward range status if `Range` itself is a forward range:

```
1 import std.range;
2
3 struct Flatten(Range) if (isInputRange!Range)
4 {
5     /* same code than before */
```

```

6
7     static if (isForwardRange!Range)
8         Flatten save() @property
9         {
10             return this;
11         }
12 }

```

The struct is enriched in two ways: first, it cannot be instantiated on a non-range. That's good because with the code from section 5.8, you could bypass the factory function and manually create a `Flatten!int`, which wouldn't do. Now, you cannot. Secondly, if the wrapped range is a forward range, then `Flatten!Range` is one also. That opens up whole new algorithms to `Flatten`, for just a quite-readable little piece of code.

You could extend the pattern in the same way by allowing `Flatten` to be a bidirectional range, but you would need to introduce a `backSubRange` member that keeps trace of the range's back state.

## 10 Template Tuple Parameters

### 10.1 Definition and Basic Properties

And now comes one of my favourite subjects: template tuple parameters. As seen in section 1 these are declared by putting a `identifier...` at the last parameter of a template. The tuple will then absorb any type, alias or literal passed to it. For this very reason (that it can bunch of types interspersed with symbols), some people consider it a mongrel addition to D templates. That is true, but the ease of use and the flexibility it gives us is in my opinion well worth the cost of a little cleanliness. D template tuples have a `.length` member (defined at compile-time, obviously), their elements can be accessed using the standard indexing syntax and they can even be sliced (the `$` symbol is aliased to the tuple length):

```

1 template DropFront(T...)
2 {
3     static if ( T.length > 0      // .length. At least 1 element
4               && is(T[0] == int)) // Indexing
5         alias T[1..$] DropFront; // Slicing
6     else
7         alias void DropFront;
8 }
9
10 alias DropFront!(int, double, string) Shortened;
11 static assert(is( Shortened[0] == double));
12 static assert(is( Shortened[1] == string));

```

You can declare a value of type 'tuple'. This value (called an expression tuple) also has a length, can be indexed and can be sliced. You can also pass it

directly to a function if the types check with a function parameter list. If you throw it into an array, it will ‘melt’ and initialize the array:

```
1 template TupleDemonstration(T...)
2 {
3     alias T TupleDemonstration;
4 }
5
6 TupleDemonstration!(string, int, double) t;
7
8 assert(t.length == 3);
9 t[0] = "abc";
10 t[1] = 1;
11 t[2] = 3.14;
12 auto t2 = t[1..$];
13 assert(t2[0] == 1)
14 assert(t2[1] == 3.14);
15
16 void foo(int i, double d) {}
17 foo(t2); // OK.
18
19 double[] array = [t2]; // see, between [ and ]
20 assert(array == [1.0, 3.14]);
```

The simplest possible tuple is already defined in Phobos in `std.tupletuple.TypeTuple`:

```
1 template TypeTuple(T...)
2 {
3     alias T TypeTuple; // It just exposes the T's
4 }
5
6 alias TypeTuple!(int, string, double) ISD;
7 static assert(is( TypeTuple!(ISD[0], ISD[1], ISD[2]) == ISD ));
```

Pure template parameter tuples are auto-flattening: they do *not* nest:

```
1 alias TypeTuple!(int, string, double) ISD;
2 alias TypeTuple!(ISD, ISD) ISDISD;
3 // ISDISD is *not* ((int, string, double), (int, string, double))
4 // It's (int, string, double, int, string, double)
5 static assert(is(ISDISD == TypeTuple!(int,string,double,int,string,double)));
```

This is both a bug and a feature. On the negative side, that condemns us to linear structures: no trees of type tuples. And since a branching structure can give rise to a linear, that would have been strictly more powerful. On the positive side, that allow us to concatenate tuples easily (and from that, to iterate easily), as you’ll see in sections 10.3 and 10.5. If you need recursive/branching structures, you can have them by using `std.typecons.Tuple` or really any

kind of struct/class template: the types are not flattened there. See for example section 29 for a fully polymorphic tree.

The last property tuples have is that they can be iterated over: use a `foreach` expression, like you would for an array. With `foreach`, you can iterate on both type tuples and expression tuples. The indexed version is also possible, but you cannot ask directly for a `ref` access to the values (but see the example below). This iteration is done at compile-time and is in fact one of the main ways to get looping at compile-time in D.

```
1 // keeping the same t and T than the previous examples.
2 string[T.length] s;
3
4 foreach(index, Type; T) // Iteration on types.
5                          // Type is a different, er, type at each position
6 {
7     static if(is(Type == double))
8         s[index] = Type.stringof;
9 }
10 assert(s == ["", "", "double"]);
11
12 void bar(T)(ref T d) { T t; d = t;}
13
14 foreach(index, value; t) // Iteration on values.
15                          // value has a different type at each position!
16 {
17     bar(t[index]); // use t[iindex], not 'value' to get a ref access
18 }
19
20 assert(t[0] == "");
21 assert(t[1] == 0);
22 assert(std.math.isnan(t[2]));
```

As values of this type can be created and named, they are *almost* first-class. They have two limitations, however:

- There is no built-in syntax for declaring a tuple. In the previous example, calling `T.stringof` returns the string "`(string,int,double)`". But you cannot write `(string,int,double) myTuple;` directly. Paradoxically, if you have a `(string,int,double)` type tuple called `T`, you *can* do `T myTuple;`.
- These tuples cannot be returned from a function. You have to wrap them in a struct. That's what `std.typecons.Tuple` offers.

### tuple, Tuple, T... and .tupleof

A common question from newcomers to D is the difference and definition between the different tuples found in the language and the standard library. I will try to explain:

*Template tuple parameters* are internal to templates. They are declared with `T...` at the last position in the parameter list. They group together a list of template parameters, be they types, values or alias. Two ‘subtypes’ are commonly used:

*Type tuples* are template tuple parameters that hold only types.

*Expression tuples* are tuples that hold only expressions. They are what you get when you declare a variable of type ‘type tuple’.

*Function parameter tuples.* You can get a function parameter type tuple from `std.traits.ParameterTypeTuple`. It’s exactly a type tuple as seen before. A value of this type can be declared and can be passed to a function with the same parameters.

The `.tupleof` property is a property of aggregate types: classes and structs. It returns an expression tuple containing the members’s values.

*Member names tuple* is a tuple of strings you get by using `__traits(members, SomeType)`. It contains all `SomeType` members’ names, as strings (including the methods, constructors, aliases and such).

`std.traits.TypeTuple` is a pre-defined template in Phobos that’s the simplest possible template holding a tuple. It’s the common D way to deal with type tuples. The name is bit of a misnomer, because it’s a standard template parameter tuple: it can hold types, but also values.

`std.typecons.Tuple` and `std.typecons.tuple` are pre-defined struct/function templates in Phobos that gives a simple syntax to manipulate tuples and return them from functions.

## 10.2 The Type of Tuples

You can get a tuple’s type by using `typeof(tuple)`, like any other D type. There are two limit cases:

**One-element tuples:** There is a difference between a tuple of one element and a lone type. You cannot initialize a standard value with a 1-element tuple. You have to extract the first (and only) element before. In the same idea, the 1-element tuple has a length and can be sliced: actions that do not make sense for a standard type.

**Zero-element tuples:** It’s possible to have an empty tuple, holding zero type, not to be confused with a uninitialized n-elements tuple or the tuple holding `void` as a type. In fact, the zero-element tuple can only have one value: its initialization value. For this reason, it’s sometimes called the Unit type.<sup>12</sup>

**void-containing tuples and empty tuples:** A type tuple may hold the `void` type, like any other D type. It ‘takes a slot’ in the tuple and a tuple holding only a `void` is *not* the empty tuple.

```
1 alias TypeTuple!(void) Void;  
2 alias TypeTuple!() Empty;
```

<sup>12</sup> Look: `bool` is a type with *two* values (`true` and `false`). `()`, the empty tuple, is the type that has only *one* value. And `void` is the type that has *no* value.

```

3 static assert( !is(Void == Empty) );
4
5 static assert(!is( TypeTuple!(int, void, string) == TypeTuple!(int, string)));

```

### 10.3 Example: Variadic Functions

Tuples are very useful to make function templates variadic (that is, accept a different number of parameters). Without restriction on the passed-in types, you will need most of the time another function template to process the arguments. A standard example for this is transforming all parameters into a `string`:

```

1 string toStrings(string sep = ", ", Args...)(Args args)
2 {
3     import std.conv:to;
4     string result;
5     foreach(index, argument; args)
6     {
7         result ~= to!string(argument);
8         if (index != args.length - 1) result ~= sep; // not for the last one
9     }
10    return result;
11 }
12
13 assert( toStrings(1, "abc", 3.14, 'a', [1,2,3])
14         == "1, abc, 3.14, a, [1,2,3]");

```

If you want to restrict the number of parameters or their types, use template constraints:

```

1 int howMany(Args...)(Args args) if (Args.length > 1 && Args.length < 10)
2 {
3     return args.length; // == Args.length
4 }

```

Imagine you have a bunch of ranges. Since they all have different types, you cannot put them in an array. And since most of them are structs, you cannot cast them to a base type, as you would for classes. So you hold them in a tuple. Then, you need to call the basic range methods on them: calling `popFront` on all of them, etc. Here is a possible way to do that:

```

1 import std.range, std.algorithm;
2
3 void popAllFronts(Ranges...)(ref Ranges ranges)
4     if(areAllRanges!Ranges)
5 {
6     foreach(index, range; ranges)
7         ranges[i].popFront; // to get a ref access
8 }
9

```

```

10 auto arr1 = [0,1,2];
11 auto arr2 = "Hello, World!";
12 auto arr3 = map!"a*a"(arr1);
13
14 popAllFronts(arr1, arr2, arr3);
15
16 assert(arr1 == [1,2]);
17 assert(arr2 == "ello, World!");
18 assert(equal( arr3, [1,4]));

```

It works for any number of ranges, that's cool. And it's checked at compile-time, you cannot pass it an `int` discretely, hoping no one will see: it's the job of `areAllRanges` to verify that. Its code is a classical example of recursion on type tuples:

```

1 template areAllRanges(Ranges...)
2 {
3     static if (Ranges.length == 0) // Base case: stop.
4         enum areAllRanges = true;
5     else static if (!isInputRange!(Ranges[0])) // Found a not-range:stop.
6         enum areAllRanges = false;
7     else // Continue the recursion
8         enum areAllRanges = areAllRanges!(Ranges[1..$]);
9 }

```

People used to languages like lisp/Scheme or Haskell will be right at home there. For the others, a little explanation might be in order:

- when you get a `typetuple`, either it's empty or it's not.
- If it's empty, then all the elements it holds are ranges and we return `true`.<sup>13</sup>
- If it's not empty, it has at least one element, which can be accessed by indexing. Let's test it: either it's a range or it's not.
- If it isn't a range, the iteration stops: not all elements are ranges, we return `false`.
- If it's a range... we have not proved anything, and need to continue.

The recursion is interesting: by defining an `areAllRanges` manifest constant, we will activate the eponymous template trick (3.1), which gets initialized to the value obtained by calling the template on a shortened tuple. With slicing, we drop the first type (it was already tested) and continue on the next one. In the end, either we exhausted the tuple (the `length == 0` case) or we find a non-range.

<sup>13</sup> You might not like it, but it's cleaner mathematically this way.



## 10.4 One-Element Tuples: Accepting Types and Alias

Sometimes it makes sense for a template to accept either a type parameter or an alias. For example, a template that returns a string representing its argument. In that case, since type parameter do not accept symbols as arguments and the same way round for alias, you're doomed to repeat yourself:

```
1 template nameOf(T)
2 {
3     enum string nameOf = T.stringof;
4 }
5
6 template nameOf(alias a)
7 {
8     Enum string nameOf = to!string(a);
9 }
10
11 static assert(nameOf!(double[]) == "double[]");
12 static assert(nameOf!(nameOf) == "nameOf");
```

Since tuples can accept both types and alias, you can use them to simplify your code a bit:

```
1 template nameOf(T...) if (T.length == 1) // restricted to one argument
2 {
3     enum string nameOf = T[0].stringof;
4 }
```

TODO: A better explanation is in order. I'm not that convinced myself.

## 10.5 Example: Inheritance Lists

THIS SECTION IS STILL A WORK IN PROGRESS:  
All this section should be rewritten. The compiler is more accepting than I thought.

Using class templates (6.1), we might want to adjust the inheritance list at compile-time. Type tuples are a nice way to it: first define a template that alias itself to a type tuple, then have the class inherit from the template:

```
1 interface I { ... }
2 interface J { ... }
3 interface K { ... }
4 interface L { ... }
5
6 class BaseA { ... }
7 class BaseB { ... }
8
```

```

9 template Inheritance(Base) if (is(Base == class))
10 {
11     static if (is(Base : BaseA))
12         alias TypeTuple!(Base, I, J, K) Inheritance;
13     else static if (is(Base : BaseB))
14         alias TypeTuple!(Base, L) Inheritance;
15     else
16         alias Base Inheritance;
17 }
18
19 // Inherits from Base
20 class MyClass : Inheritance!BaseA { ... }
21 class MyOtherClass : Inheritance!MyOtherClass { ... }

```

Here I templated `Inheritance` on the base class, but you could easily template it on a global `enum`, for example. In any case, the selection is abstracted away and the choice-making code is in one place, for you to change it easily.

There is a catch if you use it again on a derived class:

```

1 // Error! I,J and K are already listed through MyClass
2 class ErrorClass : Inheritance!(MyClass) { ... }

```

The interfaces are already listed in `MyClass` while `Inheritance` injects them again. That gets us a compilation error, because in D you cannot put an interface twice in an inheritance list. We have to do something a little more complicated: given an inheritance type list, we must eliminate all double interfaces.

Let's begin with something more simple: given a type and a type tuple, eliminate all occurrences of the type in the type tuple.

```

1 template Eliminate(Type, TargetTuple...)
2 {
3     static if (TargetTuple.length == 0) // Tuple exhausted,
4         alias TargetTuple Eliminate; // job done.
5     else static if (is(TargetTuple[0] : Type))
6         alias Eliminate!(Type, TargetTuple[1..$]) Eliminate;
7     else
8         alias
9 TypeTuple!(TargetTuple[0], Eliminate!(Type, TargetTuple[1..$])) Eliminate;
10 }
11
12 alias TypeTuple!(int,double,int,string) Target;
13 alias Eliminate!(int, Target) NoInts;
14 static assert(is( NoInts == TypeTuple!(double, string) ));

```

The only difficulty is on line 9: if the first type is not a `Type`, we have to keep it and continue the recursion:

```

1 Eliminate!(Type, Type0, Type1, Type2, ...)
2 ->
3 Type0, Eliminate!(Type, Type1, Type2, ...)

```

We cannot juxtapose types like I just did, we have to wrap them in a template. Phobos defines `TypeTuple` in `std.tupetuple` for that use.

Now that we know how to get rid of all occurrences of a type in a type tuple, we have to write a template to eliminate all duplicates. The algorithm is simple: take the first type, eliminate all occurrences of this type in the remaining type tuple. Then call the duplicate elimination anew from the resulting type tuple, while at the same time collecting the first type.

```
1 template NoDuplicates(Types...)
2 {
3     static if (Types.length == 0)
4         alias Types NoDuplicates; // No type, nothing to do.
5     else
6         alias TypeTuple!(
7             Types[0]
8             , NoDuplicates!(Eliminate!(Types[0], Types[1..$]));
9             ) NoDuplicates;
10 }
11
12 assert(is( NoDuplicates!(int,double,int,string,double)
13           == TypeTuple!(int,double,string)));
```

By the way, code to do that, also called `NoDuplicates`, is already in Phobos. It can be found in `std.tupetuple`. I found coding it again a good exercise in type tuple manipulation.

The last piece of the puzzle is to get a given class inheritance list. The `is` expression give us that by way of types specializations (A.4):

```
1 template SuperList(Class) if (is(Class = class))
2 {
3     static if (is(Class list == super))
4         alias TypeTuple!(list) SuperList;
5     else // Object
6         alias TypeTuple!() SuperList;
7 }
```

Now we are good to go: given a base class, get its inheritance list with `SuperList`. Drop the base class to keep the interfaces. Stitch with the interfaces provided by `Inheritance` and call `NoDuplicates` on it. To make things clearer, I will define many aliases in the template. To keep the use of the eponymous trick, I will defer the aliasing in another template, as seen in section 3.1.

```
1 template CheckedInheritance(Base)
2 {
3     alias CheckedImpl!(Base).Result CheckedInheritance;
4 }
5
6 template CheckedImpl(Base)
7 {
```

```

8 // Get the inheritance list, getting rid of the base class
9 static if (SuperList!(Base).length > 0)
10     alias SuperList!(Base)[1..$] InList;
11 else // Object is the only class with a zero-length inh. list
12     alias SuperList!(Base) InList;
13
14 alias TypeTuple!( InList
15                   , Inheritance!(Base)[1..$]) AllInterfaces;
16
17 alias TypeTuple!( Inheritance!(Base)[0] // base class
18                   , NoDuplicates!(AllInterfaces)) Result;
19 }
20
21 // It works!
22 class NoError : CheckedInheritance!(MyClass) { ... }

```

## 11 Operator Overloading

THIS SECTION IS STILL A WORK IN PROGRESS:

Oh yes, that's quite not finished. I'm afraid I find D syntax for operator overloading a bit heavy for my tastes. As D code in  $\text{\LaTeX}$ , it's becoming *very* cumbersome to write.

D allows users to redefine some operators to enhance readability in code. And guess what? Operator overloading is based on templates. They are described [here](#) in the docs.

### 11.1 Syntax

Table ?? gives you the operator that you can overload and which function template you must define:

Many other operators can be overloaded in D, but do not demand templates.

### 11.2 Example: Arithmetic Operators

TODO: Tell somewhere that this is possible:

```

1 Foo opBinary(string op:"+")(...) { ... }

```

The idea behind this strange way to overload operators is to allow you to redefine many operators at once with only one method. For example, take this struct wrapping a number:

```

1 struct Number(T) if (isNumeric!T)
2 {
3     T num;
4 }

```

To give it the four basic arithmetic operators with another `Number` and another `T`, you define `opBinary` for `+`, `-`, `*` and `/`. This will activate operations were `Number` is on the left. In case it's on the right, you have to define `opBinaryRight`. Since these overloading tend to use string mixins, I'll use them even though they are introduced only on section 16. The basic idea is: string mixins paste code (given as a compile-time string) where they are put.

```

1 struct Number(T) if (isNumeric!T)
2 {
3     T num;
4
5     auto opBinary(string op, U)
6         if ((op == "+" || op == "-" || op == "*" || op == "/")
7             && ((isNumeric!U) || is(U u == Number!V, V)))
8     {
9         mixin("alias typeof(a~op~"b) Result;
10            static if (isNumeric!U)
11                return Number!Result(a~op~"b);
12            else
13                return Number!Result(a~op~"b.num);");
14    }
15 }

```

`op` being a template parameter, it's usable to do compile-time constant folding: in this case the concatenation of strings to generate D code. The way the code is written, `Numbers` respect the global D promotion rules. A `Number!int` plus a `Number!double` returns a `Number!double`.

### 11.3 Special Case: `in`

### 11.4 Special Case: `cast`

## 12 Mixin Templates

Up to now, *all* the templates we have seen are instantiated in the same scope than their declaration. Mixin templates have a different behaviour: the code they hold is placed upon instantiation *right at the call site*. They are thus used in a completely different way than other templates.

### 12.1 Syntax

To distinguish standard templates from mixin templates, the latter have slightly different syntax. Here is how they are declared and called:

```

1 /* Declaration */
2 mixin template NewFunctionality(T,U) { ... }
3
4
5 /* Instantiation */
6 class MyClass(T,U,V)

```

```

7 | {
8 |     mixin NewFunctionality!(U,V);
9 |
10 |     ...
11 | }

```

As you can see, you put `mixin` before the declaration and `mixin` before the instantiation call. All other templates niceties (constraints, default values, ...) are still there for your perusal. Symbols lookup is done in the local scope and the resulting code is included where the call was made, therefore injecting new functionality.

As far as I know, there is no special syntax for function, class and struct templates to be mixin templates. You will have to wrap them in a standard `template` declaration. In the same idea, there is no notion of eponymous trick with mixin templates: there is no question of how to give access to the template's content, since the template is cracked open for you and its very content put in your code.

**TODO: Test for mixin T foo(T)(T t) return t;**

By the way, you *cannot* mix a standard template in. It used to be the case, but it's not possible anymore. Now mixin templates and non-mixin ones are strictly separated cousins.

## 12.2 Mixing Code In

What good are these cousins of the templates we've seen so far? They give you a nice way to place parameterized implementation inside a class or a struct. Once more, templates are a way to reduce boilerplate code. If some piece of code appears in different places in your code (for example, in structs, where there is no inheritance to avoid code duplication), you should look for a way to put it in a mixin template.

Also, you can put small functionalities in mixin templates, giving client code access to them to chose how they want to build their types.

Note that the code you place inside a mixin template doesn't have to make sense by itself (it can refer to `this` or any not-yet-defined symbols). It just has to be syntactically correct D code.

For example, remember the operator overloading code we saw in section 11? Here is a mixin containing concatenating functionality:

```

1 | mixin template Concatenate()
2 | {
3 |     Tuple!(This, U) opBinary(string op, this This, U)(U u)
4 |     if (op == "~")
5 |     {
6 |         return tuple(this, u);
7 |     }
8 |
9 |     Tuple!(U, This) opBinaryRight(string op, this This, U)(U u)
10 |    if (op == "~")
11 |    {
12 |        return tuple(u, this);

```

```

13     }
14 }

```

As you can see, it uses `this`, even though there is no struct or class in sight. It's used like this, to give concatenation (as tuples) ability to a struct:

```

1 struct S
2 {
3     /* some code */
4
5     mixin Concatenate;
6 }
7
8 S s,t,u;
9
10 auto result = s ~ t ~ u;
11 assert(result == tuple(s, tuple(t,u)));

```

In this particular case, we should test for tuples already containing the current type and flatten them, so as get `tuple(s,t,u)`.

The idea to take back home is: the concatenation code is written once. It is then an offered functionality for any client scope (type) that want it. It could easily have been arithmetic operations, `cast` operations or new methods like `log`, `register`, new members or whatever else. Build you own set of mixins and use them freely. And remember they are not limited to classes and structs: you can also use them in functions, module scopes, other templates...

### Limitations

Mixin templates inject code at the local scope. They cannot add an `invariant` clause in a class, or `in/out` clauses in a function. They can be injected into an `invariant/in/out` clause.

## 13 opDispatch

### 13.1 Syntax

`opDispatch` is a sort of operator overloading (it's in the same place in the [online documentation](#)) that deals with members calls (methods or value members). Its definition is the same than an operator:

```

1 ... opDispatch(string name, Args)(Arg arg)
2 ... opDispatch(string name, Args...)(Args args)

```

The usual template constraints can be used: constraints on `name`, constraints on the arguments.

When a type has an `opDispatch` method and a member call is done without finding a defined member, the call is dispatched to `opDispatch` with the invoked name as a string.

```

1 struct Dispatcher
2 {
3     int foo(int i) { return i*i;}
4     string opDispatch(string name, T...)(T t)
5     {
6         return "Dispatch activated: " ~ name ~ ":" ~ T.stringof;
7     }
8 }
9
10 Dispatcher d;
11
12 auto i = d.foo(1); // compiler finds foo, calls foo.
13 auto s1 = d.register("abc"); // no register member -> opDispatch activated;
14 assert(s1 == "Dispatch activated: register:string");
15
16 auto s2 = d.empty; // no empty member, no argument.
17 assert(s2 == "Dispatch activated: empty:()");

```

Once `opDispatch` has the name called and the arguments, it's up to you to decide what to do: calling free functions, calling other methods or using the compile-time string to generate new code (see section 16 on string mixins).

Since string mixins really go hand in hand with `opDispatch` I'll use them even though I haven't introduced them right now. The executive summary is: they paste D code (given as a compile-time string) where they are called. There.

## 13.2 Getters and Setters

For example, suppose you have a bunch of members, all private and want client code to access them through good ol' `setXXX/getXXX` methods. Only, you do not want to write all these methods by yourself. You lucky you, `opDispatch` can help you.

```

1 class GetSet
2 {
3     private int i;
4     private int j;
5     private double d;
6     private string theString;
7
8     auto opDispatch(string name)() // no arg version -> getter
9     if (name.length > 3 && name[0..3] == "get")
10    {
11        enum string member = name[3..$]; // "getXXX" -> "XXX"
12        // We test if "XXX" exists here: ie if is(typeof(this.XXX)) is true
13        static if (mixin("is(typeof(this." ~ name ~ ")"))
14            mixin("return " ~ name ~ ";");
15        else
16            static assert(0, "GetSet Error: no member called " ~ name);
17    }

```



```

18
19 auto opDispatch(string name, Arg)(Arg arg) // setter
20 if (name.length > 3 && name[0..3] == "set")
21 {
22     enum string member = name[3..$]; // "setXXX" -> "XXX"
23     // We test if "name" can be assigned to. this.name = Arg.init
24     static if (__traits(compiles,
25         mixin("this." ~ name ~ " = Arg.init;")))
26         mixin("return " ~ name ~ ";");
27     else
28         static assert(0, "GetSet Error: no member called " ~ name);
29 }
30 }
31
32 auto gs = new GetSet();
33 gs.seti(3);
34 auto i = gs.geti;
35 assert(i == 3);
36
37 gs.settheString("abc");
38 writeln(gs.gettheString); // "abc"

```

Nifty, eh? This could be a bit better by dealing with the capitalization of the first letter: `getTheString`, but this is good enough for now. Even better, you could put this code in a mixin template to give this get/set capacity to any struct or class (see section 12).

### 13.3 Wrapper Templates

THIS SECTION IS STILL A WORK IN PROGRESS:  
The idea is to put some functionality inside before (and possibly, after, the dispatching code.

We've seen how to inject code with mixin templates (12) or use template class inheritance to modify you classes' code (10.5). We've also seen how you can define a wrapper struct around a range to expose a new iteration scheme for its element (5.8). All these idioms are way to modify pre-existing code.

But what you want to put a logging functionality around a predefined struct, so that any method call is logged? For class, you can inherit from the class and defined a subclass with new, modified, methods. But you have to do that 'by hand', so to speak. And for a struct, you're out of luck.

But, templates can come to the rescue, with a bit of `opDispatch` magic.

TODO: Finish this.

- put Type wrapped into a Logger struct. - get Type.tupleof - call typeof() on this. - opDispatch? Test if wrapped.foo() is legal. If yes, call

## 14 Templates All the Way Up

THIS SECTION IS STILL A WORK IN PROGRESS:

In this section, you'll see: Double-decker templates, Curried templates and how to separate two type tuples.

## 15 `__FILE__` and `__LINE__`

THIS SECTION IS STILL A WORK IN PROGRESS:

This section needs some heavy testing. My D config was broken when I wrote this part.

In section 3.5, we've seen that template parameters can have default values. There are also two special, reserved, symbols that are defined in D: `__FILE__` and `__LINE__`. They are used in standard (non-mixin) templates, but their behaviour will remind you of mixins: when instantiated, they get replaced by strings containing the file name and the line in the file of the *instantiation call site*. Yes, it's a sort of two-way dialogue: module `a.d` defines template `T`. Module `b.d` asks for a `T` instantiation. This instantiation is done in module `a.d`, but will line and filename taken from `b.d`!

They are mostly declared like this:

```
1 struct Unique(T, string file, size_t line)
2 {
3     enum size_t l = line;
4     enum string f = file;
5     T t;
6 }
7
8 auto unique(T, string file = __FILE__, size_t line = __LINE__)(T t)
9 {
10     return Unique!(T, file, line)(t);
11 }
```

As `Unique`'s name suggests, this is a way to obtain unique instantiations. Except if you call the very same template twice in the same line of your file, this pretty much guarantee your instantiation will be the only one. Remember that template arguments become part of the template scope name when instantiation is done (2).

```
1 // file thefile.d
2 module thefile;
3
4 auto u = unique(1); // Unique!(int, "thefile.d", 4)
5
```

```

6 auto v = unique(1); // Unique!(int, "thefile.d", 6)
7
8 static assert(!is( typeof(v) == typeof(u) ))

```

Even though `u` and `v` are declared the same way, they have different types. Apart from *one-of-a-kind* types, this is also useful for debugging: you can use the strings in error messages:

```

1 auto flatten(Range, file == __FILE__, line == __LINE__)(Range range)
2 {
3     static if (rank!Range == 0)
4         static assert(0, "File: " ~ file ~ " at line: " ~ line
5                         ~ ", flatten called with a rank-0 type: "
6                         ~ Range.stringof);
7     else static if (rank!Range == 1)
8         return range;
9     else
10        return Flatten!(Range)(range);
11 }

```

And here is a little gift:

```

1 template Debug(alias toTest, file == __FILE__, line == __LINE__)
2 {
3     template With(Args...)
4     {
5         static if (is( toTest!Args ))
6             alias toTest!Args With;
7         else
8             static assert(0, "Error: " ~ to!string(toTest)
9                             ~ " called withs arguments: "
10                             ~ Args.stringof);
11     }
12
13     /* Usage */
14     Debug!(templateToBeTested).With!(Arguments);

```

That way, no need to modify your beautiful templates.

**TODO:** Test that.

## Part III

# Around Templates: Other Compile-Time Tools

There is more to compile-time metaprogramming in D than *just* templates. This part will describe the most common tools: string mixins (16), compile-time function evaluation (17) and `__traits` (18), as seen in relation with templates. For the good news is: they are all interoperable. String mixins are wonderful to inject code in your templates, compile-time-evaluable functions can act as template parameters and can be templated. And, best of best, templated compile-time functions can return strings which can in turn be mixed-in... in your templates. Come and see, it's fun!

## 16 String Mixins

String mixins put D code where they are called, just before compilation. Once injected, the code is *bona fide* D code, like any other. Code is manipulated as strings, hence the name.

### 16.1 Syntax

The syntax is slightly different from mixin templates (12):

```
1 mixin("some code as a string");
```

You must take care not to forget the parenthesis. String mixins are a purely compile-time tool, so the string must also be determined at compile-time.

### 16.2 Mixing Code In, With Templates

Of course, just injecting predefined code is a bit boring:

```
1 mixin("int i = 3;"); // Do not forget the two semicolons
2                      // one for the mixed-in code,
3                      // one for the mixin() call.
4 i++;
5 assert(i == 4);
```

There is no interest in that compared to directly writing standard D code. The fun begins with D powerful constant folding ability: in D, strings can be concatenated at compile-time. That's where string mixins meet templates: templates can produce strings at compile-time and can get strings as parameters. You already saw that in section 11 on operator overloading and section 13 on `opDispatch`, since I couldn't help doing a bit of foreshadowing.

Now, imagine for example wanting a template that generates structs for you. You want to be able to name the structs as you wish. Say we would like the usage to look like that:

```

1 module mine;
2 import named;
3
4 mixin(Named!"First"); // creates struct First { ... }
5 mixin(Named!"Second"); // and struct Second { ... }
6
7 First f1, f2;
8 Second s1;
9
10 assert(is( typeof(s1) == mine.First));

```

Here comes the generating code:

```

1 module named;
2
3 template Named(string name)
4 {
5     enum string Named = "struct " ~ name ~ " { "
6                         ~ "+ some code +"
7                         ~ " }";
8 }
9
10 /* For example, name == "First" ->
11    struct First { + some code + }
12 */

```

In this case, the string is assembled inside the template during instantiation, exposed through the eponymous trick and then mixed in where you want it. Note that the string is generated in the module containing `Named`, but that `First` and `Second` are defined exactly where the `mixin()` call is. If you use the `mixin` in different modules, this will define as many different structs, all named the same way. This might be exactly what you want, or not.

To get the same struct in different modules, the code must be organized a bit differently: the structs must be generated in the template module (for example):

```

1 module named;
2
3 template Named(string name)
4 {
5     alias NamedImpl!(name).result Named;
6 }
7
8 template NamedImpl(string name)
9 {
10     enum string Named = "struct " ~ name ~ " { "
11                       ~ "+ some code +"
12                       ~ " }";
13     mixin(Named);

```

```

14     mixin("alias " ~ name ~ " result;");
15 }

```

```

1 module mine;
2 import named;
3
4 named!"First" f1, f2;
5 named!"Second" s1;

```

Usage is a different, as you can see. In this case, `First` is generated inside `NamedImpl` and exposed through an alias (this particular alias statement is itself generated by a string mixin). In fact, the entire code could be put in the mixin:

```

1 module named;
2
3 template Named(string name)
4 {
5     alias NamedImpl!(name).result Named;
6 }
7
8 template NamedImpl(string name)
9 {
10  mixin("struct " ~ name ~ " {"
11         ~ "/* some code */"
12         ~ " }\n"
13         ~ "alias " ~ name ~ " result;");
14 }

```

Here is an example using the ternary `?:` operator to do some compile-time selection of code, similar to what can be done with `static if` (3.3):

```

1 enum GetSet { no, yes}
2
3 struct S(GetSet getset = GetSet.no, T)
4 {
5     enum priv = "private T value;\n"
6                 ~ "T get() @property { return value;}\n"
7                 ~ "void set(T _value) { value = _value;}";
8
9     enum pub = "T value;";
10
11     mixin( (getset == GetSet.yes) ? priv : pub);
12 }
13
14 S!(GetSet.yes, int) gs;
15
16 /* Generates:
17
18 struct S!(GetSet.yes, int)

```

```

19 | {
20 |     private int value;
21 |     int get() @property { return value;}
22 |     void set(int _value) { value = _value;}
23 | }
24 | */
25 |
26 | gs.set(1);
27 | assert( gs.get == 1);

```

## 16.3 Limitations

Code crafting is still a bit awkward, because I haven't introduced CTFE yet (see 17). So we are limited to simple concatenation for now: looping for example is possible with templates, but far easier with CTFE. Even then, it's already wonderfully powerful: you can craft D code with some 'holes' (types, names, whatever) that will be completed by a template instantiation and then mixed in elsewhere. You can create other any kind of D code with that.

You can put `mixin()` expressions almost were you want to, but...

**TODO:** Test the limits: inside static if expressions, for example

### Escaping strings

One usual problem with manipulating D code as string is how to deal with strings in code? You must escape them. Either use `\"` to create string quotes, a bit like was done in section 4.1 to generate the error message for `select`. Or you can put strings between `q{` and `}`.

## 17 Compile-Time Function Evaluation

### 17.1 Evaluation at Compile-Time

Compile-Time Function Evaluation (from now on, CTFE) is an extension of the constant-folding that's done during compilation in D code: if you can calculate `1 + 2 + 3*4` at compile-time, why not extend it to whole functions evaluation? I'll call evaluable at compile-time functions CTE functions from now on.

It's a very hot topic in D right now and the reference compiler has advanced by leaps and bounds in 2011. The limits to what can be done with CTE functions are pushed farther away at each new release. All the `foreach`, `while`, `if/then/else` statements, arrays manipulation, struct manipulation, function manipulation... are there. You can even do pointer arithmetics! When I began this document (DMD 2.055), the limitations were mostly: no classes and no exceptions (and so, no `enforce`). This was changed with DMD 2.057, allowing the manipulation of classes at compile-time.

In fact danger lies the other way round: it's easy to forget that CTE functions must also be standard, runtime, functions. Remember that some actions only make sense at compile-time or with compile-time initialized constants: indexing on tuples for example:

## 17.2 `__ctfe`

THIS SECTION IS STILL A WORK IN PROGRESS:  
Write something on this new functionality, which enables testing inside a function whether we are at compile-time or runtime.

## 17.3 Templates and CTFE

THIS SECTION IS STILL A WORK IN PROGRESS:  
Some juicy examples should be added.

That means: you can feed compile-time constants to your classical D function and its code will be evaluated at compile-time. As far as templates are concerned, this means that function return values can be used as template parameters and as `enum` initializers:

1

Template functions can very well give rise to functions evaluated at compile-time:

1

## 17.4 Templates and CTFE and String Mixins, oh my!

And the fireworks is when you mix (!) that with string mixins: code can be generated by functions, giving access to almost the entire D language to craft it. This code can be mixed in templates to produce what you want. And, to close the loop: the function returning the code-as-string can itself be a template, using another template parameters as its own parameters.

Concretely, here is the getting-setting code from section 16.2, reloaded:

```
1 enum GetSet { no, yes}
2
3 string priv(string type, string index)
4 {
5     return
6     "private "~type~"value"~index~";\n"
7     ~ type~" get"~index~"() @property { return value"~index~";}\n"
8     ~ "void set"~index~"("~type~" _value) { value"~index~" = _value;}";
9 }
10
11 string pub(string type, string index)
12 {
13     return type ~ "value" ~ index ~ ";";
14 }
```



```

15
16 string GenerateS(GetSet getset = GetSet.no, T... )()
17 {
18     string result;
19     foreach(index, Type; T)
20         static if (getset == GetSet.yes)
21             result ~= priv(Type.stringof, to!string(index));
22         else
23             result ~= pub(Type.stringof, to!string(index));
24     return result;
25 }
26
27 struct S(GetSet getset = GetSet.no, T...)
28 {
29     mixin(GenerateS!(getset,T));
30 }
31
32 S!(GetSet.yes, int, string, int) gs;
33 /* Generates:
34
35 struct S!(GetSet.yes, int, string, int)
36 {
37     private int value0;
38     int get0() @property { return value0;}
39     void set0(int _value) { value0 = _value;}
40
41     private string value1;
42     string get1() @property { return value1;}
43     void set1(string _value) { value1 = _value;}
44
45     private int value2;
46     int get2() @property { return value2;}
47     void set2(int _value) { value2 = _value;}
48 }
49 */
50
51 gs.set1("abc");
52 assert(gs.get1 == "abc");

```

This code is much more powerful than the one we saw in section 16.2: the number of types is flexible, and an entire set of getters/setters is generated when asked to. All this is done by simply plugging `string`-returning functions together, and a bit of looping by way of a compile-time `foreach`.

## 17.5 Simple String Interpolation

All this play with the concatenating operator (`~`) is becoming a drag. We should write a string interpolation function, evaluable at compile-time of course, to help us in our task. Here is how I want to use it:

```

1 alias interpolate!"struct #0 { #1 value; #0[#2] children;}" makeTree;
2
3 enum string intTree = makeTree("IntTree", "int", 2);
4 enum string doubleTree = makeTree("DoubleTree", "double", "");
5
6 assert(intTree
7     == "struct IntTree { int value; IntTree[2] children;}");
8 assert(doubleTree
9     == "struct DoubleTree { double value; IntTree[] children;}");

```

As you can see, the string to be interpolated is passed as a template parameter. Placeholders use a character normally not found in D code: `#`. The  $n^{th}$  parameter is `#n`, starting from 0. As a concession to practicality, a lone `#` is considered equivalent to `#0`. Args to be put into the string are passed as standard (non-template) parameters and can be of any type.

```

1 template interpolate(string code)
2 {
3     string interpolate(Args...)(Args args) {
4         string[] stringified;
5         foreach(index, arg; args) stringified ~= to!string(arg);
6
7         string result;
8         int i;
9         int zero = to!int('0');
10
11         while (i < code.length) {
12             if (code[i] == '#') {
13                 int j = 1;
14                 int index;
15                 while (i+j < code.length
16                     && to!int(code[i+j])-zero >= 0
17                     && to!int(code[i+j])-zero <= 9)
18                 {
19                     index = index*10 + to!int(code[i+j])-zero;
20                     ++j;
21                 }
22
23                 result ~= stringified[index];
24                 i += j;
25             }
26             else {
27                 result ~= code[i];
28                 ++i;
29             }
30         }
31
32         return result;

```

```

33     }
34 }

```

TODO: The syntax could be extended somewhat: inserting multiple strings, inserting a range of strings, all arguments to the end.

## 17.6 Example: extending `std.functional.binaryFun`

THIS SECTION IS STILL A WORK IN PROGRESS:

This one is dear to my heart. Mapping  $n$  ranges in parallel is one of the first things that I wanted to do with ranges, for examples to create ranges of structs with constructor taking more than one parameter.

Phobos has two really interesting templates: `std.functional.unaryFun` and `std.functional.binaryFun`.

TODO: Explain that this aims to extend that to n-args functions.

```

1 bool isaz(char c) {
2     return c >= 'a' && c <= 'z';
3 }
4
5 bool isAZ(char c) {
6     return c >= 'A' && c <= 'Z';
7 }
8
9 bool isNotLetter(char c) {
10    return !isaz(c) && !isAZ(c);
11 }
12
13 int letternum(char c) {
14    return to!int(c) - to!int('a') + 1;
15 }
16
17 int arity(string s) {
18    if (s.length == 0) return 0;
19
20    int arity;
21    string padded = " " ~ s ~ " ";
22    foreach(i, c; padded[0..$-2])
23        if (isaz(padded[i+1])
24            && isNotLetter(padded[i])
25            && isNotLetter(padded[i+2]))
26            arity = letternum(padded[i+1]) > arity ?
27                    letternum(padded[i+1])
28                    : arity;
29    return arity;
30 }
31

```

```

32 string templateTypes(int arit) {
33     if (arit == 0) return "";
34     if (arit == 1) return "A";
35
36     string result;
37     foreach(i; 0..arit)
38         result ~= "ABCDEFGHJKLMNOPQRSTUVWXYZ"[i] ~ ", ";
39
40     return result[0..$-2];
41 }
42
43 string params(int arit) {
44     if (arit == 0) return "";
45     if (arit == 1) return "A a";
46
47     string result;
48     foreach(i; 0..arit)
49         result ~= "ABCDEFGHJKLMNOPQRSTUVWXYZ"[i]
50             ~ " " ~ "abcdefghijklmnopqrstuvwxyz"[i]
51             ~ ", ";
52
53     return result[0..$-2];
54 }
55
56 string naryFunBody(string code, int arit) {
57     return interpolate!"auto ref naryFun(#0)(#1) { return #2;}"
58         (templateTypes(arit), params(arit), code);
59 }
60
61 template naryFun(string code, int arit = arity(code))
62 {
63     mixin(naryFunBody(code, arit));
64 }

```

## 18 \_\_traits

THIS SECTION IS STILL A WORK IN PROGRESS:

I should maybe add an example about how to get all parents from a class type. Also, iterating on all members of a type, or extracting types and names from an aggregate (a module, a class or a struct).

The general `__traits` syntax can be found online [here](#). Traits are basically another compile-time introspection tool, complementary to the `is` expression (see [Appendix A](#)). Most of time, `__traits` will return `true` or `false` for simple type-introspection questions (is this type or symbol an abstract class, or a final function?). As D is wont to do, these questions are sometimes ones you could

ask using `is` or template constraints, but sometimes not. What’s interesting is that you can do some introspection on types, but also on symbols or expressions.

## 18.1 Yes/No Questions with `__traits`

Seeing how this is a document on templates and that we have already seen many introspection tools, here is a quick list of what yes/no questions you can ask which can or *cannot* be tested with `is`.<sup>14</sup>

Question	Doable with other tools?
<code>isArithmetic</code>	Yes
<code>isAssociativeArray</code>	Yes
<code>isFloating</code>	Yes
<code>isIntegral</code>	Yes
<code>isScalar</code>	Yes
<code>isStaticArray</code>	Yes
<code>isUnsigned</code>	Yes
<code>isAbstractClass</code>	No
<code>isFinalClass</code>	No
<code>isVirtualFunction</code>	No
<code>isAbstractFunction</code>	No
<code>isFinalFunction</code>	No
<code>isStaticFunction</code>	No
<code>isRef</code>	No
<code>isOut</code>	No
<code>isLazy</code>	No
<code>hasMember</code>	No (Yes?)
<code>isSame</code>	No
<code>compiles</code>	Yes (in a way)

Table 1: Comparison between `__traits` and other introspection tools

These can all be quite useful in your code, but I’ll shy away from them since they are not heavily template-related. More interesting in my opinion is using `__traits` to get new information about a type. These are really different from other introspection tools and I will deal with them in more detail right now.

## 18.2 `__traits(identifier, XXX)`

`identifier` gives you a symbol’s name as a string. This is quite interesting, since some symbols are what I’d call *active*: for example, if `foo` is a function, `foo.stringof` will try to first call `foo` and the `.stringof` will fail. Also, `.stringof`, though eminently useful, sometimes returns strangely formatted strings. `identifier` is much more well-behaved.

Let’s get back one of the very first template in this doc, `NameOf` on page ?? . Initially, it was coded like this:

<sup>14</sup>As with any other affirmation in this document, readers should feel free to prove me wrong. That shouldn’t be too difficult.

```

1 template NameOf(alias a)
2 {
3     enum string name = a.stringof; // enum: manifest constant
4                                     // determined at compile-time
5 }

```

But, this fail for functions:

```

1 int foo(int i, int j) { return i+j;}
2
3 auto name = NameOf!foo; // Error, 'foo' must be called with 2 arguments

```

It's much better to use `__traits` (also, the eponymous trick (3.1)):

```

1 template nameOf(alias a)
2 {
3     enum string nameOf = __traits(identifier, a);
4 }
5
6 int foo(int i, int j) { return i+j;}
7
8 enum name = nameOf!foo; // name is available at compile-time
9
10 assert(name == "foo");

```

Note that this works for many (all?) kinds of symbols: template names, class names, even modules: auto

```

1 import std.typecons;
2
3 enum name2 = nameOf!(nameOf); // "nameOf"
4 enum name3 = nameOf!(std.typecons); // "typecons"

```

### 18.3 getMember

In a nutshell, `__traits(getMember, name, "member")` will give you direct access to `name.member`. This is the real member: you can get its value (if any), set it anew, etc. Any D construct with members is OK as a **name**. If you wonder why the aggregate is called directly by its own name whereas the member needs a string, it's because the aggregate is a valid symbol (it exists by itself), when the member's name has no existence outside the aggregate (or even worse, may refer to another, unrelated, construct).

#### Aggregates

I'll use *aggregate* as a catch-all term for any D construct that has members.

Structs and classes are obvious aggregates, but it's interesting to keep in mind that templates too have members (remember section 2? A template is a named, parameterized scope). So all `__traits` calls shown in this section can be used on templates. That's interesting to keep in mind. Even more interesting, to my eyes, is that *modules* are also aggregates even though they are not first-class citizens in D-land. You'll see examples of this in the following sections.

## 18.4 allMembers

This one is cool. Given an aggregate name, `__traits(allMembers, aggregate)` will return a tuple of string literals, each of which is the name of a member. For a class, the parent classes' members are also included. Built-in properties (like `.sizeof`) are not included in that list. Note that I did say 'tuple': it's a bit more powerful than an array, because it can be iterated over at compile-time.

The names are not repeated for overloads, but see the next section for a way to get the overloads.

```
1 class MyClass
2 {
3     int i;
4
5     this() { i = 0;}
6     this(int j) { i = j;}
7     ~this() { }
8
9     void foo() { ++i;}
10    int foo(int j) { return i+j;}
11 }
12
13 // Put in an array for a more human-readable printing
14 enum myMembers = [__traits(allMembers, MyClass)];
15
16 // See "i" and "foo" in the middle of standard class members
17 // "foo" appears only once, despite it being overloaded.
18 assert(myMembers == ["i", "__ctor", "__dtor", "foo", "toString",
19                      "toHash", "opCmp", "opEquals", "Monitor", "factory"]);
```

So the above code is a nice way to get members, both fields (like `i`) and methods (like `foo`). In case you wonder what `"__ctor"` and `"__dtor"` are, it's the internal D name for constructors and destructors. But it's perfectly usable in your code! For structs, the list far less cluttered since they only get the constructor and destructor's names and `opAssign`, the assignment operator (`=`).

Since this trait returns strings, it can be plugged directly into `getMember`. See a bit farther down a way to get a nice list of all members and their types.

Now, let's ramp things up a bit: what about class and struct templates? Let's try this out:

```

1 class MyClass(T)
2 {
3     T field;
4 }
5
6 assert([__traits(allMembers, MyClass)] == ["MyClass"]);

```

Oh, what happened? Remember from [Part I](#) that struct, classes and functions templates are just syntactic sugar for the 'real' syntax:

```

1 template MyClass(T)
2 {
3     class MyClass
4     {
5         T field;
6     }
7 }

```

So, MyClass is just the name of the external template, whose one and only member is...the MyClass class. So all is well. If you instantiate the template, it functions as you might expect:

```

1 assert([__traits(allMembers, MyClass!int)]
2     == ["field", "toString", "toHash", "opCmp",
3         "opEquals", "Monitor", "factory"]);

```

If you remember one of the very first use we saw for templates in [2](#), that is as a named, parameterize scope, this can give rise to interesting introspection:

```

1 template Temp(A, B)
2 {
3     A a;
4     B foo(A a, B b) { return b;}
5     int i;
6     alias A     AType;
7     alias A[B] AAType;
8 }
9
10 assert([__traits(allMembers, Temp)]
11     == ["a", "foo", "i", "AType", "AAType"]);
12 assert([__traits(allMembers, Temp!(double,string))]
13     == ["a","foo", "i", "AType", "AAType"]);

```

As you can see, you also get aliases. By the way, this is true for structs and templates also...

Another fun fact is that D modules are amenable to `__traits`'s calls:



```

1 // file a.d
2 module a;
3
4 int foo(int i) { return i;}
5 class MyClass { }

```

```

1 // file b.d
2 module b;
3
4 import a; // bring 'a' as a symbol in the module's scope.
5 import std.algorithm;
6
7 enum aMembers = __traits(allMembers, a);
8 enum algos = __traits(allMembers, std.algorithm); // huuuge list of names
9
10 enum myself = __traits(allMembers, b); // auto-introspection!
11
12 void main()
13 {
14     assert(aMembers == ["object", "foo", "MyClass"]);
15     assert(myself == ["object", "a", "std", "main"]);
16 }

```

In the previous code, you see that among `a` members, there is "object" (the implicit `object.d` module imported by the runtime). And since auto-introspection is possible, we also get `b`'s members. There, there is also "a", as it's imported into `b`, and "std", the global package that shows here whenever you import a `std.*` module. It would be easy to imagine a template that recursively explore the members, find the modules and try to recurse into them to get a complete import-tree with a template. Halas "std" blocks that, since the package itself do not have a member.<sup>15</sup>

As for the other aggregates, you get the aliases also, and the unit tests defined in the module (for those of you curious about it, they are named `__unittestXXX` where XXX is a number. Their type is more or less `void delegate()`).<sup>16</sup>

What's the point of inspecting a module, you say? Well, first that was just for fun and to see if I could duplicate a module or create a struct with an equivalent members list (all forwarding to the module's own members). But the real deal for me was when using string mixins to generate some type. If the user uses the mixin in its own module, it could create conflicts with already-existing names. So I searched for a way for a mixin template to inspect the module it's currently being instantiated in.

<sup>15</sup> If someone finds a way to determine with a template that `b` imports `std.algorithm`, I'd be delighted to see how it's done!

<sup>16</sup> I didn't try static constructors in modules. Don't hesitate to play with them and tell me what happens.

## 18.5 derivedMembers

Really it's the same as above, except that you do not get a class' parents members, only the class' own members.

## 18.6 getOverloads

Given:

- an aggregate name or an instance of that aggregate
- a member name as a string

Then, `__traits(getOverloads, name, "member")` will give you a tuple of all local overloads of `name.member`. By 'local', I mean that for classes, you do not get the parents classes overloads. There is a difference between using `getOverloads` on a type and on an instance: in the former case, you get ..., well I don't know exactly what it is you get but it's a bit akin to static members. On an instance, you get an expression that is the exact member overload. The [documentation](#) is misleading: it says `getOverloads` returns an array, but that cannot be the case, as all overloads must have a different type. What's cool is that, by using `"__ctor"`, you also get a direct access to a type's constructor overloads. That can be quite handy in some cases.

## 18.7 Getting All Members, Even Overloaded Ones

Now, if you're like me, the urge to mix `allMembers` (which returns the members' names without overloads) and `getOverloads` (which returns the overload of *one* member) is quite great. So let's do that.

First, a bit of machinery: I want the members to be described by a name and a type. Let's create a struct holder, templated of course:

**TODO: Getting qualified names would be better.**

```
1 struct Member(string n, T)
2 {
3     enum name = n; // for external access
4     alias T Type;
5 }
```

Given a member name, we'll need a template that delivers the associated Member:

```
1 template MakeMember(alias member)
2 {
3     alias Member!(nameOf!member, typeof(member)) MakeMember;
4 }
```

Now, given an aggregate name and a member name (as a string, since these do not exist by themselves), we want a list of `Member` structs holding all the information:

```

1  template Overloads(alias a, string member)
2  {
3      alias OverloadsImpl!(__traits(getOverloads,a, member)).result Overloads;
4  }
5
6  template OverloadsImpl(A...)
7  {
8      alias staticMap!(MakeNameAndType,A) result;
9  }

```

It's a two-stage template as explained in sections 3.1 (eponymous templates) and 14 (to get the overloads expansion as a tuple parameter). `staticMap` is explained in section ??.

Now, that already works:

```

1  class MyClass
2  {
3      int i;
4
5      this() { i = 0;}
6      this(int j) { i = j;}
7      ~this() { }
8
9      void foo(int j) { ++i;}
10     int foo(int j, int k = 0) { return i+j;}
11 }
12
13 void main()
14 {
15     alias Overloads!(MyClass, "foo") o;
16
17     /*
18     prints:
19     foo of type: void(int j)
20     foo of type: int(int j, int k = 0)
21     */
22     foreach(elem; o)
23         writeln(elem.name, " of type: ", elem.Type.stringof);
24 }

```

We indeed got two `Member` instances, one for each overload. Each `Member` holds the name "foo" and the overload's type. Except, there is a catch: for a field, there are no overloads. We need to pay attention to that in `Overloads`:

```

1  template Overloads(alias a, string member)
2  {
3      static if (__traits(compiles, __traits(getOverloads, a, member)))
4          alias OverloadsImpl!(__traits(getOverloads,a, member)).result

```

```

5         Overloads;
6     else
7         alias TypeTuple!(Member!(member, typeof(__traits(getMember, a, member))))
8         Overloads;
9 }

```

Line 3 is an example of the *very handy* `__traits(compiles, someExpression)` construct alluded to in section 18.1. Line 7 can be a bit daunting, but it just creates the one-element tuple `__traits` should have returned.

The last step is to get this for all members of a given aggregate. It's quite easy:

```

1 template GetOverloads(alias a)
2 {
3     template GetOverloads(string member)
4     {
5         alias Overloads!(a, member) GetOverloads;
6     }
7 }
8
9 template AllMembers(alias a)
10 {
11     alias staticMap!(GetOverloads!(a), __traits(allMembers, a)) AllMembers;
12 }

```

The strange `GetOverloads` two-stage construction is just a way to map it more easily on line 11. So, this was quite long to explain, but it works nicely:

```

1 void main()
2 {
3     alias AllMembers!MyClass o;
4     /*
5     prints:
6     i of type: int
7     __ctor of type: MyClass()
8     __ctor of type: MyClass(int j)
9     __dtor of type: void()
10    foo of type: void(int j)
11    foo of type: int(int j, int k = 0)
12    toString of type: string()
13    toHash of type: uint()
14    opCmp of type: int(Object o)
15    opEquals of type: bool(Object o)
16    opEquals of type: bool(Object lhs, Object rhs)
17    factory of type: Object(string classname)
18    */
19    foreach(elem; o)
20        writeln(elem.name, " of type: ", elem.Type.stringof);
21 }

```

That's cool, every member and overload is accounted for: the two constructors are there, the destructor also, and of course `i`, of type `int`.

TODO: Example: store all members in a hashtable or a polymorphic association list. As a mixin, to be put inside types to enable runtime reflection? (`a.send("someMethod", args)`, `a.setInstanceVariable("i",5)`)

## 18.8 Traits TODO

- `getVirtualFunctions` (returns an array of the virtual overloads)
- `parent` (get the parent directly). Example: recurse upwards.

Example: iterating on members, extracting types from an aggregate.

## 19 Wrapping it all Together

THIS SECTION IS STILL A WORK IN PROGRESS:

Here is what I want to show: using traits to crack a type open, templated CT functions to generate strings to be mixed-in, cloning a type.

### 19.1 Cracking Open a Type

## Part IV

# Examples

This part will present various examples showing what can be done with D templates, be it type manipulation, code generation or language extension ... Most examples are code snippets that were useful to me at one time or another. I hope they will be useful to you too.

### Contributors welcome!

Even more than for other parts, I welcome any short and synthetic example of what can be done with templates. Do not hesitate to chime in and push some code in this doc!

## 20 Type Sorcery

One of the most fundamental use of templates is type sorcery: type creation, type manipulation, etc. D being a statically type language, all of your creations will have a defined type. Sometimes, these can be cumbersome to write or manipulate. Templates can help you in this.

### 20.1 Mapping, Filtering and Folding Types

As we saw in section 10, template tuple parameters can hold type tuples (that's even their original role). Since these can be iterated, indexed or sliced, they are ideal candidates to some standard iteration algorithms. As for ranges, you can map another template on type tuples, filter the types you want to extract or fold (reduce) them into another type.

### And non-types?

What about acting on expression tuples? You can do that too. Even though this section is called *Type Sorcery*, all templates here can work on expression tuples too. Do not feel limited.

#### 20.1.1 Mapping on Type Tuples

Mapping a type tuple is simply applying another (unary) template on all types of a type tuple. Phobos already defines `staticMap` in `std.tupetuple`, but it's a nice exercise to code it again. We want a template that takes another template name (as an `alias` parameter), say `Wrapper`, and a `typetuple` (`T0`, `T1`, `T2`, ..., `Tn`) and returns `Wrapper!T0`, `Wrapper!T1`, `Wrapper!T2`, ..., `Wrapper!Tn`.

```
1 template staticMap(alias M, T...)
2 {
3     static if (T.length == 0) // End of sequence
4         alias TypeTuple!() staticMap; // stop there
```

```

5     else
6         alias TypeTuple!(M!(T[0]), staticMap!(M, T[1..$])) staticMap;
7 }

```

We use the auto-flattening of type tuples to aggregate the results into a unique tuple. Notice how indexing and slicing make for a not-so-complicated piece of code.

Even a simple template such as this one can have great uses:

- Getting rid of all qualifiers in a type list, by mapping `std.traits.Unqual` on the types.
- Generating a huge quantity of types by using a typetuple-returning mapper (see below).
- Given a bunch of function types, getting their return types or parameter typetuples.

### 20.1.2 Example: Testing a Function

The seconde use in the previous section's list can be useful to unit-test a part of your code. Suppose you have a templated function that is supposed to work on any built-in type. You do not need to generate all possible combinations of types. Just use `staticMap` to generate them for you:

```

1  /**
2   * Given a type T, generates all qualified versions of T
3   * that you find interesting (eleven versions all in all).
4   */
5  template Qualified(T)
6  {
7      alias TypeTuple!(
8          T, const(T), immutable(T), shared(T),
9          T[],const(T) [],immutable(T) [], shared(T) [],
10         const(T[]), immutable(T[]), shared(T[])
11         ) Qualified;
12 }
13
14 // All 16 built-in types you're interested in.
15 alias TypeTuple!(
16     bool,
17     ubyte,byte,
18     ushort,short,
19     uint,int,
20     ulong,long,
21     float,double,real,
22     char,wchar,dchar
23     ) ValueTypes;
24
25 // Bang, 11*16 types generated.
26 alias staticMap!(Qualified,ValueTypes) QualifiedTypes;

```

```

27 |
28 | // If you're really a pervert (note that there will be duplicates)
29 | alias staticMap!(Qualified, QualifiedTypes) DoublyQualifiedTypes;

```

Now, if your function is supposed to work on all the generated qualified types, just test it:

```

1 void myFun(T)(T t) { ... }
2
3 /* Here we generate the qualified types, see above */
4 (...)
5 temp
6 late test(alias fun)
7 {
8     void on(T... )()
9     {
10         foreach(Type; T)
11             static if (!__traits(compiles, fun(Type.init)))
12                 pragma(msg, "Bad testing combination: "
13                     ~ fun.stringof ~ " and " ~ Type.stringof);
14     }
15 }
16
17 unittest
18 {
19     test!(myFun).on!(QualifiedTypes);
20 }

```

### 20.1.3 Filtering Type Tuples

You can search for and extract some types from a tuple, using a predicate to chose which type (or more generally, which tuple element) you want to keep. A predicate in this particular case means 'a template that, when given a type as argument, will return either **true** or **false**'. The test done on the tuple element can be as complicated as you want, particularly using **is()** expressions (see [Appendix A](#)).

That gives us the following code:

```

1 template staticFilter(alias Pred, T...)
2 {
3     static if (T.length == 0) // End of sequence
4         alias TypeTuple!() staticFilter;
5     else static if (Pred!(T[0])
6         alias TypeTuple!(T[0], staticFilter!(Pred, T[1..$])) staticFilter;
7     else
8         alias TypeTuple!(staticFilter!(Pred, T[1..$])) staticFilter;
9 }

```

Using **staticFilter** is quite simple. Let's get integral types from a tuple, by way of **std.traits.isIntegral**:



```

1 alias TypeTuple!(int, double, float, string, byte, bool, float, void) Types;
2
3 alias staticFilter!(isIntegral, Types) OnlyIntegrals;
4
5 static assert(is(OnlyIntegrals == TypeTuple!(int, byte, bool)));

```

But that is, admittedly, pretty boring stuff. Though useful from time to time, it's quite rare for someone to be given a pure type tuple like this. A much more common use for the likes of `staticFilter` is when creating complex types.

#### 20.1.4 Example: building a Graph

As a first example, imagine you have a `Graph(Node, Edge)` struct, templated on the nodes (vertex) and edges types (themselves templated). When you create a `Graph` with a factory function (5.2), it would be nice to be able to mix nodes and edges in a natural way. That is, given `graph`, `node` and `edge` functions that do the obvious thing, you want to authorize calls like:

```

1 /**
2  * Automatically creates a graph
3  *   - with four nodes labelled "A", "B", "C", "D", holding a double,
4  *   - with nodes linking "A" to "B", "D" to "A" and "D" to "C".
5  */
6 auto g = graph(node("A", 3.14159), node("B", 1.0),
7               edge("A", "B"),
8               node("C", 2.71828), node("D", 0.0),
9               edge("D", "A"), edge("D", "C"));

```

This allows the user building her `Graph` to create nodes and edges between these nodes in a natural way (as opposed to, say, batch-building all nodes and then adding edges between them). But, as a library writer, that means your `graph` factory function has the following signature:

```

1 auto graph(NodesOrEdges...)(NodesOrEdges args)
2 if (/* sanity check test on NodesOrEdges */)

```

Both the sanity check performed by the template constraint (8) and the building code inside `graph` can be quite complicated. `staticFilter` helps by separating the arguments between nodes and edges. Without extending this example too much, say we have at our disposal the following predicate templates:

```

1 template isNode(N) { /* true iff N is a Node!(LabelType, ValueType) */ }
2 template isEdge(E) { /* true iff E is an Edge!(LabelType) */ }
3
4 template isNodeOrEdge(T)
5 {
6     static if (isNode!T || isEdge!T)
7         enum isNodeOrEdge = true;

```

```

8     else
9         enum isNodeOrEdge = false;
10 }

```

And let's suppose also all *bona fide* Nodes and Edges have `.LabelType/.ValueType` members exposing their inner types (as shown in section 3.2).

Then, getting all nodes and edges is easy:

```

1 alias staticFilter!(isNode, NodesOrEdges) Nodes;
2 alias staticFilter!(isEdge, NodesOrEdges) Edges;

```

This is where things get interesting: obtaining the edges and nodes types is just a first building block. Now `graph` must check at a minimum the following elements:

1. All arguments must be nodes or edges.
2. Is there at least *one* node in the list?
3. If yes, do all nodes have the same `LabelType` and the same `ValueType`, or, at least, is there a common type between all the labels' types and another one for the values stored in the nodes?
4. Do edges' `LabelTypes` have a common type? (Note that there can be zero edges).
5. Do the edges labels have the correct type to refer to the provided nodes?

Note that *all* these checks are done only on types and thus can be done at compile-time, thereby ensuring a pretty solid static check on the graph built. What cannot be done in this way is verifying during compilation that edges do indeed refer to existing nodes.

Let's use what we have seen until now to create the `GraphCheck` template, before seeing another `staticFilter` example:

```

1 import std.traits: CommonType;
2
3 template GraphCheck(NodesOrEdges...)
4 {
5     enum GraphCheck = GraphCheckImpl!(NodesOrEdges).result;
6 }
7
8 template GraphCheckImpl(NodesOrEdges...)
9 {
10     alias staticFilter!(isNode, NodesOrEdges) Nodes;
11     alias staticFilter!(isEdge, NodesOrEdges) Edges;
12
13     // 1. All arguments must be nodes or edges
14     static if (Nodes.length + Edges.length != NodesOrEdges.length)
15         static assert(0, "Some args are not nodes or edges.");
16

```

```

17 // 2. There must be at least one node
18 static if (Nodes.length == 0)
19 static assert(0, "You must provide at least one node.");
20
21 // 3. Is there a common type for the nodes' labels and values?
22 // First step: extracting labels and values
23 template GetLabel(T) if (isNode!T || isEdge!T)
24 {
25     alias T.LabelType GetLabel;
26 }
27
28 template GetValue(T) if (isNode!T)
29 {
30     alias T.ValueType GetValue;
31 }
32
33 alias staticMap!(GetLabel, Nodes) NodesLabels;
34 alias staticMap!(GetValue, Nodes) NodesValues;
35
36 static if (is(CommonType!(NodesLabels) == void)) // no common type
37 static assert(0, "The nodes do not have all the same label type.");
38
39 static if (is(CommonType!(NodesValues) == void))
40 static assert(0, "The nodes do not have all the same value type.");
41
42 // 4. Same for edges
43 alias staticMap!(GetLabel, Edges) EdgesLabels;
44
45 static if (is(CommonType!(EdgesLabels) == void))
46 static assert(0, "The edges do not have all the same label type.");
47
48 // 5. Edges - Node compatibility
49 static if(!is(CommonType!NodesLabels == CommonType!EdgesLabels))
50     static assert(0, "Nodes and edges do not have the same label type.");
51
52 enum result = true;
53 }

```

This is one huge template, but `staticFilter` sits square in the middle and greatly simplifies the code. Now, `graph` signature is simply:

```

1 auto graph(NodesOrEdges...)(NodesOrEdges args) if (GraphCheck!NodesOrEdges)
2 { ... }

```

### 20.1.5 Folding Type Tuples

TODO: Examples to give: biggest type in the lot, sorting types?

## 20.2 Zipping Types, Interspersing Types, Crossing Types

TODO: Determine if this subsection is really useful. Is there any short and telling example?

```
1 /**
2  Helper template. Given T0, T1, T2, ..., Tn, Tn+1, ... T2n, will returns
3  the interleaving of the first part with the second part: T0, Tn+1, T1, Tn+2, ... Tn, T2n
4  It's fragile: no test, etc. A better way to do this would be as a two-steps template: Interlea
5  */
6 template Interleave(T...)
7 {
8     static if (T.length > 1)
9         alias TypeTuple!(T[0], T[$/2], Interleave!(T[1..$/2], T[$/2+1 .. $])) Interleave;
10    else
11        alias T Interleave;
12 }
```

## 21 Relational Algebra

Inspiration for this example comes from [This blog article](#).

TODO: Extracting from a tuple: project, select. Also, natural/inner/outer join, cartesian product. And intersection/union/difference. `rename!( "oldField", "newField")`. Databases are just dynamic arrays of tuples..

## 22 Cloning, sort of

TODO: (Elsewhere) creating a class from a struct?

## 23 Generic Maker Function

Like this:

```
1 class C
2 {
3     int i, j;
4
5     this(int _i) { i = _i; j = _i;}
6     this(int _i, int _j) { i = _i; j = _j;}
7 }
8
9 alias make!C makeC;
10
11 auto theCs = map!makeC([0,1,2,3,4]);
12 auto theCs2 = map!makeC(zip([0,1,2,3,4],
13                             [4,3,2,1,0]));
```

## 24 Recording Successive States

From Andrej Mitrovic.

```
1 import std.stdio;
2 import std.traits;
3
4 struct Shape
5 {
6     int x, y;
7     void foo(int val) { x += val; }
8     int bar(int val) { y += val; return y; }
9 }
10
11 struct RecorderImpl(T)
12 {
13     T t;
14     T[] t_states;
15
16     this(T t)
17     {
18         this.t = t;
19         t_states ~= t;
20     }
21
22     auto opDispatch(string method, Args...)(Args args)
23     {
24         static if (mixin("is(ReturnType!(t." ~ method ~ ") == void)"))
25         {
26             mixin("t." ~ method ~ "(args);");
27             t_states ~= t;
28         }
29         else
30         {
31             mixin("auto result = t." ~ method ~ "(args);");
32             t_states ~= t;
33             return result;
34         }
35     }
36
37     auto opIndex(size_t index)
38     {
39         assert(index < t_states.length);
40         return t_states[index];
41     }
42
43     auto opSlice(size_t lowerBound, size_t upperBound)
44     {
45         return t_states[lowerBound..upperBound];
46     }
```

```

47 |
48 |     auto opSlice()
49 |     {
50 |         return t_states[];
51 |     }
52 | }
53 |
54 | auto Recorder(T)(T t)
55 | {
56 |     return RecorderImpl!T(t);
57 | }
58 |
59 | void main()
60 | {
61 |     auto shape = Recorder(Shape(0, 0));
62 |
63 |     shape.foo(5);
64 |     shape.bar(5);
65 |
66 |     writeln(shape[0]);
67 |     writeln(shape[]);
68 | }

```

## 25 Fields

From Jacob's Carlborg Orange.

**TODO:** Test `typeof(s.tupleof)`

```

1  /**
2   * Evaluates to an array of strings containing the names of the fields in the given type
3   */
4  template fieldsOf (T)
5  {
6      const fieldsOf = fieldsOfImpl!(T, 0);
7  }
8
9  /**
10 * Implementation for fieldsOf
11 *
12 * Returns: an array of strings containing the names of the fields in the given type
13 */
14 template fieldsOfImpl (T, size_t i)
15 {
16     static if (T.tupleof.length == 0)
17         const fieldsOfImpl = [""];
18
19     else static if (T.tupleof.length - 1 == i)
20         const fieldsOfImpl = [T.tupleof[i].stringof[1 + T.stringof.length + 2 .. $]];
21

```

```

22         else
23             const fieldsOfImpl = T.tupleof[i].stringof[1 + T.stringof.length + 2 .. $] ~ f
24     }

```

## 26 Extending an enum

```

1  string EnumDefAsString(T)() if (is(T == enum)) {
2      string result = "";
3      foreach (e; __traits(allMembers, T)) {
4          result ~= e ~ " = T." ~ e ~ ",";
5      }
6      return result;
7  }
8
9  template ExtendEnum(T, string s) if (is(T == enum) &&
10 is(typeof({mixin("enum a{"~s~"}");}))) {
11      mixin("enum ExtendEnum {" ~
12          EnumDefAsString!T() ~ s ~
13          "}");
14  }
15
16 unittest {
17     enum bar {
18         a = 1,
19         b = 7,
20         c = 19
21     }
22
23     import std.tuple;
24
25     alias ExtendEnum!(bar, q{ // Usage example here.
26         d = 25
27     }) bar2;
28
29     foreach (i, e; __traits(allMembers, bar2)) {
30         static assert( e == TypeTuple!("a", "b", "c", "d")[i] );
31     }
32
33     assert( bar2.a == bar.a );
34     assert( bar2.b == bar.b );
35     assert( bar2.c == bar.c );
36     assert( bar2.d == 25 );
37
38     static assert(!is(typeof( ExtendEnum!(int, "a"))));
39     static assert(!is(typeof( ExtendEnum!(bar, "25"))));
40 }
41
42 //(Simen Kjaeraas)

```

## 27 Static Switching

**TODO:** What, no compile-time switch? Let's create one. Example of: tuples, type filtering (in constraints), recursion, etc.

```
1 template staticSwitch(List...) // List[0] is the value commanding the switching
2                               // It can be a type or a symbol.
3 {
4     static if (List.length == 1) // No slot left: error
5         static assert(0, "StaticSwitch: no match for " ~ List[0].stringof);
6     else static if (List.length == 2) // One slot left: default case
7         enum staticSwitch = List[1];
8     else static if (is(List[0] == List[1]) // Comparison on types
9                     || ( !is(List[0])      // Comparison on values
10                        && !is(List[1])
11                        && is(typeof(List[0] == List[1]))
12                        && (List[0] == List[1])))
13         enum staticSwitch = List[2];
14     else
15         enum staticSwitch = staticSwitch!(List[0], List[3..$]);
16 }
```

## 28 Gobble

```
1 struct Gobbler(T...)
2 {
3     T store;
4     Gobbler!(T, string, U) opBinary(string op, U)(U u) if (op == "~")
5     {
6         return Gobbler!(T, string, U)(store, op, u);
7     }
8 }
9
10 Gobbler!() gobble() { return Gobbler!()(); }
```

## 29 A Polymorphic Tree

## 30 Polymorphic Association Lists

Usage: a bit like Lua tables: structs, classes (you can put anonymous functions in them?), namespaces. Also, maybe to add metadata to a type?

```
1 template Half(T...)
2 {
3     static if (T.length <= 1)
4         alias TypeTuple!() Half;
5     else
```



```

6      alias TypeTuple!(T[0], Half!(T[2..$])) Half;
7  }
8
9  struct AList(T...)
10 {
11     static if (T.length >= 2 && T.length % 2 == 0)
12         alias Half!T Keys;
13     else static if (T.length >= 2 && T.length % 2 == 1)
14         alias Half!(T[0..$-1]) Keys;
15     else
16         alias TypeTuple!() Keys;
17
18     static if (T.length >= 2)
19         alias Half!(T[1..$]) Values;
20     else
21         alias TypeTuple!() Values;
22
23     template at(alias a)
24     {
25         static if ((staticIndexOf!(a, Keys) == -1) && (T.length % 2 == 1)) // key not found, b
26             enum at = T[$-1]; // default value
27         else static if ((staticIndexOf!(a, Keys) == -1) && (T.length % 2 == 0))
28             static assert(0, "AList: no key equal to " ~ a.stringof);
29         else //static if (Keys[staticIndexOf!(a, Keys)] == a)
30             enum at = Values[staticIndexOf!(a, Keys)];
31     }
32 }
33
34 alias AList!( 1,      "abc"
35               , 2,      'd'
36               , 3,      "def"
37               , "foo", 3.14
38               ,        "Default") al;
39
40 writeln("Keys: ", al.Keys);
41 writeln("Values: ", al.Values);
42 writeln("at!1: ", al.at!(1));
43 writeln("at!2: ", al.at!(2));
44 writeln("at!\\"foo\\": ", al.at!("foo"));
45 writeln("Default: ", al.at!4);

```

## 31 Expression Templates

## 32 Wrapping a Function

Making it accept a tuple, for example.

```

1 template tuplify(alias fun)
2 {

```

```

3 auto tuplify(T...)(Tuple!T tup)
4 {
5     return fun(tup.expand);
6 }
7 }

```

Another interesting (and much more complicated) example is `juxtapose`.

### 33 Mapping $n$ ranges in parallel

```

1 // Very easy to do, now:
2 auto nmap(alias fun, R...)(R ranges) if (allSatisfy!(isInputRange, R))
3 {
4     return map!(tuplify!fun)(zip(ranges));
5 }

```

More complicated: `std.algorithm.map` accepts to take more than one function as template argument. In that case, the functions are all mapped in parallel on the range, internally using `std.functional.adjoin`. Here we can extend `nmap` to accept  $n$  functions in parallel too. There is a first difficulty:

```

1 auto nmap(fun..., R...)(R ranges) if (allSatisfy!(isInputRange, R))
2 { ...

```

See the problem? Tuples must be the last parameter of a template: there can be only one. Double-stage templates come to the rescue:

```

1 template nmap(fun...) if (fun.length >= 1)
2 {
3     auto nmap(R...)(R ranges) if (allSatisfy!(isInputRange, R))
4     {...}
5 }

```

Final code:

```

1 template nmap(fun...) if (fun.length >= 1)
2 {
3     auto nmap(R...)(R ranges) if (allSatisfy!(isInputRange, R))
4     {
5         alias adjoin!(staticMap!(tuplify, fun)) _fun;
6         return map!(_fun)(zip(ranges));
7     }
8 }

```

Give an example with `max`, it works!

And here is the  $n$ -ranges version of `std.algorithm.filter`:

```

1 auto nfilter(alias fun, R...)(R ranges) if (allSatisfy!(isInputRange, R))
2 {
3     return filter!(tupleify!fun)(zip(ranges));
4 }

```

## 34 Statically-Checked Writeln

## 35 Extending a Class

There is regularly a wish in the D community for something called Universal Function Call Syntax (UFCS): the automatic transformation of `a.foo(b)` into `foo(a,b)` when `a` has no member called `foo` and there *is* a free function called `foo` in the local scope. This already works for arrays (hence, for strings) but not for other types.

There is no way to get that in D for built-in types except by hacking the compiler, but for user-defined types, you can call templates to the rescue.

`opDispatch` can be used to forward to an external free function. A call `this.method(a,b)` becomes `method(this,a,b)`.

```

1 mixin template Forwarder
2 {
3     auto opDispatch(string name, Args...)(Args args)
4     {
5         mixin("return " ~ name ~ "(args);");
6     }
7 }

```

In D, a void `return` clause is legal:

```

1 return;
2 // or return void;

```

So if `name(this,a,b)` is a `void`-returning function, all is OK.

The main limitation of this trick is that it doesn't work across modules boundaries. Too bad.

## 36 Pattern Matching With Functions

## 37 Generating a Switch for Tuples

Case 0:, etc.

## 38 Tuples as Sequences

### 38.1 Mapping on Tuples

### 38.2 Filtering Tuples

```
1 (1, "abc", 2, "def", 3.14)
2 ->
3 ((1,2),("abc","def"),(3,14))
```

# Appendices

## A A Crash Course on the `is(...)` Expression

### A.1 General Syntax

The `is(...)` expression gives you some compile-time introspection on types (and, as a side-effect, on D expressions). It's described [here](#) in the D website. This expression has a quirky syntax, but the basic use is very simple and it's very useful in conjunction with `static if` (see section 3.3 and template constraints (see section 8)). The common syntaxes are:

```
1 is( Type (optional identifier) )
2 is( Type (optional identifier) : OtherType,
3     (optional template parameters list) )
4 is( Type (optional identifier) == OtherType,
5     (optional template parameters list) )
```

If what's inside the parenthesis is valid (see below), `is()` returns `true` at compile-time, else it returns `false`.

### A.2 `is(Type)`

Let's begin with the very first syntax: if `Type` is a valid D type in the scope of the `is` expression, `is()` returns `true`. As a bonus, inside a `static if`, the optional `identifier` becomes an alias for the type. For

```
1 template CanBeInstantiatedWith(alias templateName, Types...)
2 {
3     // is templateName!(Types) a valid type?
4     static if (is( templateName!(Types) ResultType ))
5         // here you can use ResultType (== templateName!(Types))
6         alias ResultType CanBeInstantiatedWith;
7     else
8         alias void CanBeInstantiatedWith;
9 }
```

Note that the previous code was done with templates in mind, but it is quite robust: if you pass as `templateName` something that's not a template name (a function name, for example), the `is` will see `templateName!(Types)` has no valid type and will return `false`. `CanBeInstantiatedWith` will correctly be set to `void` and your program does not crash.

#### Testing for an alias

Sometimes you do not know if the template argument you received is a type or an alias (for example, when dealing with tuples elements). In that case, you can use `!is(symbol)` as a test. If it really is an alias and not a type, this will return `true`.

An interesting use for this form of `is`, is testing whether or not some D code is valid. Consider: D blocks are seen as delegates by the compiler (Their type is `void delegate()`). Using this in conjunction with `typeof` let you test the validity of a block statement: if `some code` is valid, `typeof({ some code }())` (note the `()` at the end of the delegate to 'activate' it) has a real D type and `is` will return true.

Let's put this to some use. Imagine you have a function template `fun` and some arguments, but you do not know if `fun` can be called with this particular bunch of arguments. If it's a common case in your code, you should abstract it as a template. Let's call it `validCall` and make it a function template also, to easily use it with the arguments:

```

1 bool validCall(alias fun, Args...)(Args args)
2 {
3     static if (is( typeof({ /* code to test */
4                           fun(args);
5                           /* end of code to test */
6                           }())))
7         return true;
8     else
9         return false;
10 }
11
12 // Usage:
13 T add(T)(T a, T b) { return a+b;}
14
15 assert( validCall!add(1, 2.3)); // generates add!(double)
16 assert(!validCall!add(1, "abc")); // no template instantiation possible
17
18 string conc(A,B)(A a, B b) { return to!string(a) ~ to!string(b);}
19
20 assert( validCall!conc(1, "abc")); // conc!(int, string) is OK.
21 assert(!validCall!conc(1) ); // no 1-argument version for conc
22
23 struct S {}
24
25 assert(!validCall!S(1, 2.3); // S is not callable

```

Note that the tested code is simply `'fun(args);'`. That is, there is no condition on `fun`'s type: it could be a function, a delegate or even a struct or class with `opCall` defined. There are basically two ways `fun(args);` can be invalid code: either `fun` is not callable as a function, or it is callable, but `args` are not valid arguments.

By the way, fun as it may be to use this trick, D provides you with a cleaner way to test for valid compilation:

```

1 __traits(compiles, { /* some code */ })

```

`__traits` is another of D numerous Swiss Army knife constructs. You can find the `compiles` documentation [here](#). Section 18 is dedicated to it.

### A.3 `is(Type : AnotherType)` and `is(Type == AnotherType)`

The two other basic forms of `is` return true if `Type` can be implicitly converted to (is derived from) `AnotherType` and if `Type` is exactly `AnotherType`, respectively. I find them most interesting in their more complex form, with a list of template parameters afterwards. In this case, the template parameters act a bit as type variables in an equation. Let me explain:

```
1 is(Type identifier == SomeComplexTypeDependingOnUAndV, U, V)
```

really means: ‘excuse me, Mr. D Compiler, but is `Type` perchance some complex type depending on `U` and `V` for some `U` and `V`? If yes, please give me those.’  
For

```
1 template ArrayElement(T)
2 {
3     // is T an array of U, for some U?
4     static if (is(T t : U[], U))
5         alias U ArrayElement; // U can be used, let's expose it
6     else
7         alias void ArrayElement;
8 }
9
10 template isAssociativeArray(AA)
11 {
12     static if (is( AA aa == Value[Key], Value, Key))
13         /* code here can use Value and Key,
14         they have been deduced by the compiler. */
15     else
16         /* AA is not an associative array
17         Value and Key are not defined. */
18 }
```

Strangely, you can only use it with the `is(Type identifier, ...)` syntax: you *must* have `identifier`. The good new is, the complex types being inspected can be templated types and the parameter list can be any template parameter: not only types, but integral values, .... For example, suppose you do what everybody does when encountering D templates: you create a templated n-dimensional vector type.

```
1 struct Vector(Type, int dim) { ... }
```

If you did not expose `Type` and `dim` (as aliases for example, as seen in sections 3.2 and 5.3), you can use `is` to extract them for you:

```
1 Vector!( ?, ?) myVec;
2 // is myVec a vector of ints, of any dimension?
3 static if (is(typeof(myVec) mv == Vector!(int, dim), dim))
4
5 // is it a 1-dimensional vector?
6 static if (is(typeof(myVec) mv == Vector!(T, 1), T))
```

### is( A != B)?

No, sorry, this doesn't exist. Use `!is(A == B)`. But beware this will also fire if `A` or `B` are not legal types (which makes sense: if `A` is not defined, then by definition it cannot be equal to `B`). If necessary, you can use `is(A) && is(B) && !is(A == B)`.

### is A a supertype to B?

Hey, `is(MyType : SuperType)` is good to know if `MyType` is a subtype to `SuperType`. How do I ask if `MyType` is a supertype to `SubType`? Easy, just use `is(SubType : MyType)`.

For me, the main limitation is that template tuple parameters are not accepted. Too bad. See, imagine you use `std.typecons.Tuple` a lot. At one point, you need a template to test if something is a `Tuple!(T...)` for some `T` which can be 0 or more types. Though luck, `is` is a bit of a letdown there, as you cannot do:

```
1 template isTuple(T)
2 {
3     static if (is(T tup == Tuple!(InnerTypes), InnerTypes...)
4 (...))
```

But sometimes D channels its inner perl and, lo! There is more than one way to do it! You can use IFTI (see 4.3) and our good friend the `is(typeof(...))` expression there. You can also use `__traits`, depending on you mood, but since this appendix is specifically on `is`:

```
1 template isTuple(T)
2 {
3     static if (is(typeof({
4         void tupleTester(InnerTypes...)(Tuple!(InnerTypes) tup) {}
5         T.init possibleTuple;
6         tupleTester(possibleTuple);
7         }()))
8         enum bool isTuple = true;
9     else
10         enum bool isTuple = false;
11 }
```

Line 4 defines the function template `tupleTester`, that only accepts `Tuples` as arguments (even though it does nothing with them). We create something of type `T` on line 5, using the `.init` property inherent in all D types, and try to call `tupleTester` with it. If `T` is indeed a `Tuple` this entire block statement is valid, the resulting delegate call indeed has a type and `is` returns `true`.

There are two things to note here: first, `isTuple` works for any templated type called `Tuple`, not only `std.typecons.Tuple`. If you want to restrict it,



change `tupleTester` definition. Secondly, we do not get access to the inner types this way. For `std.typecons.Tuple` it's not really a problem, as they can be accessed with the `someTuple.Types` alias, but still...

By the way, the template parameter list elements can themselves use the `A : B` or `A == B` syntax:

```
1 static if (is( T t == A!(U,V), U : SomeClass!W, V == int[n], W, int n))
```

This will be OK if `T` is indeed an `A` instantiated with an `U` and a `V`, themselves verifying that this `U` is derived from `SomeClass!W` for some `W` type and that `V` is a static array of `ints` of length `n` to be determined (and possibly used afterwards). In the `if` branch of the `static if` `U`, `V`, `W` and `n` are all defined.

## A.4 Type Specializations

There is a last thing to know about `is`: with the `is(Type (identifier) == Something)` version, `Something` can also be a type specialization, one of the following D keywords: `function`, `delegate`, `return`, `struct`, `enum`, `union`, `class`, `interface`, `super`, `const`, `immutable` or `shared`. The condition is satisfied if `Type` is one of those (except for `super` and `return`, see below). `identifier` then becomes an alias for some property of `Type`, as described in table 2.

Specialization	Satisfied if	<code>identifier</code> becomes
<code>function</code>	Type is a function	The function parameters type tuple
<code>delegate</code>	Type is a delegate	The delegate function type
<code>return</code>	Type is a function or a delegate	The return type
<code>struct</code>	Type is a struct	The struct type
<code>enum</code>	Type is an enum	The enum base type
<code>union</code>	Type is an union	The union type
<code>class</code>	Type is a class	The class type
<code>interface</code>	Type is an interface	The interface type
<code>super</code>	Type is a class	The type tuple (Base Class, Interfaces)
<code>const</code>	Type is const	The type
<code>immutable</code>	Type is immutable	The type
<code>shared</code>	Type is shared	The type

Table 2: Effect of type specializations in `is`

Let's put that to some use: we want a factory template that will create a new struct or a new class, given its name as a template parameter:

```
1 import std.algorithm;
2
3 template make(alias aggregate)
4     if (is(typeof(aggregate) == class )
5         || is(typeof(aggregate) == struct))
6 {
7     auto make(Args...)(Args args)
```

```

8      {
9          alias typeof(aggregate) A;
10         static if (is(A a == class))
11             return new A(args);
12         else
13             return A(args);
14     }
15 }
16
17 struct S {int i;}
18 class C {int i; this(int ii) { i = ii;}}
19
20 auto array = [0,1,2,3];
21
22 auto structRange = map!( make!S )(array);
23 auto classRange  = map!( make!C )(array);
24
25 assert(equal(structRange, [S(0), S(1), S(2), S(3)]));
26 assert(equal(structRange, [new C(0), new C(1), new C(2), new C(3)]));

```

You can find another example of this kind of `is` in section 6, with the duplicator template.

## Index

- anonymous
  - class templates, 48
  - classes, 48
  - delegates, 27
  - function templates, 27
  - functions, 26
- array
  - base element type, 18
- arrays, 17, 19, 25
  - and expression tuples, 60
  - UFCS, 107
- Boost, 6
- C++, 6, 8, 21, 22, 57
  - C++0x, 54
  - CRTP, 51
- C#, 12
- class templates, 65
- compile-time
  - constant (`enum`), 10
  - constant folding, 69, 79
  - introspection, 16, 17, 53, 57, 109
    - with `__traits`, 84
  - iteration, 63
  - looping, 61
  - string concatenation, 76
- Compile-Time Function Evaluation, *see* CTFE, 79
- CTFE, 9, 58, 79
  - and templates, 80
  - limitations, 79
- Curiously Recurring Template Pattern, 51
- D Zen, 57
- eponymous trick, 13, 16, 19, 20, 64, 67
  - functions storage classes, 28
  - string mixins, 77
- example
  - `__FILE__` and `__LINE__`
    - debugging, 75
    - unique types, 74
  - anonymous class templates, 48
  - anonymous function template, 27
  - arithmetic operators, 69
  - `auto` return, 24
  - class template
    - invariant, 47
  - class templates and type tuples, 65
  - compile-time iteration, 63
  - CTFE, 80
  - enabling/disabling code, 16, 31
  - enum template, 52
  - `enum`-based policy, 31
  - eponymous trick, 13
  - exposing through an alias, 14
  - expression tuple, 60
  - factory function, 35, 43, 45
  - function composition, 19
  - function template, 23
  - IFTI, 45
  - `in` and `out` clauses, 29
  - inner member alias, 43
  - inner struct, 40, 41
    - templated, 41
  - `is` expression, 109, 111
    - types specialization, 113
    - types specializations, 67
  - `is(typeof())`, 110, 112
  - mixin templates, 70
  - nested `static ifs`, 16
  - nested templates instantiations, 49
  - `opCall`, `()`, 41
  - `opDispatch`, 71, 72
  - parameterized base class, 48
  - predicate template
    - on type tuples, 64
  - predicate templates, 54, 58
    - member testing, 57
    - operator testing, 58
  - range, 43, 58
  - range templates, 54
  - recursion, 45
  - returning a function, 26
  - secondary template, 14
  - `static if`, 45
    - enabling/disabling code, 29
  - storage class, 28
    - function parameters, 28
  - string mixins, 72, 76–78
  - struct template, 34, 35, 43
    - inner alias parameters, 36

- template constraints, 53, 54
- template definition, 9
- template instantiation, 11
- template recursion, 17
- template tuple parameter, 59
- templated constructor, 40
- templated method, 37, 38
- templated methods, 41
- templates constraints limitations, 55
- templates specialization, 21
- tuples
  - and ranges, 63
- type tuple, 59
- variadic functions, 63

factory function, 45

function templates
 

- anonymous function templates, 27
- wrapping a function
  - memoize, 30, 31

generics, 12, 23

Haskell, 64

idiom
 

- enabling/disabling code, 16, 31
- factory function, 36
- parameterized base class, 49
- policy, 31
- template specialization, 22
- two templates, 14, 67
- wrapper template, 73

IFTI, 24, 45, 112
 

- for constructors, 40

Implicit Function Template Instantiation, 24

introspection
 

- with `__traits`, 84

`is` expression, 16, 53, 57, 66
 

- compared to constraints, 58
- in a constraint, 57
- limitation, 112
- type specialization, 28
- types specializations, 67, 113

Java, 12

lisp, 64

mantra, 25–27, 41, 48, 52

minted (L<sup>A</sup>T<sub>E</sub>X package), 7

mixin templates
 

- limitations, 71

module, 9

operator
 

- arithmetic, 69
- concatenation, (~), 70
- global scope operator (`.`), 20
- `opCall`, (`()`), 20, 33, 41
- overloading, 68

perl
 

- inner perl in D, 112

Phobos, 6, 14, 16–18, 22, 23, 25, 30, 35–38, 42, 53, 54, 60–62, 67, 83, 94–96, 106, 112, 113

predicate templates, 57, 64
 

- `areAllRanges`, 64
- for ranges, 54

range, 17, 26, 36, 38, 42, 43, 54
 

- base element type, 18
- higher-level ranges, 16
- members, 42
- of ranges, 18, 42, 43

rank, 17, 25
 

- definition, 17
- for ranges, 18
- of ranges, 45

recursion, 17, 19, 25, 37, 45, 64, 66
 

- `areAllRanges`, 64

Scheme, 64

scope, 10, 15
 

- client, 71
- constraint, 53
- global scope operator, 20
- local scope, 71, 107, 109
- mixin templates, 69
  - injecting functionality, 70
  - symbols lookup, 70
- multiple templates instantiations, 12
- named, 11
- outside the scope of this document, 20
- scopes are not first-class in D, 25
- `static if`, 15

- template declaration, 11
- template instantiation, 12
- template scope, 74
- top-level, 9
- `static assert`, 33
- `static assert`, 17
- `static if`, 19, 57
  - enabling/disabling code
    - `invariant` clause, 47
  - functions storage classes, 28
  - nested, 16, 28, 33, 64
  - recursion, 33
- `std`
  - algorithm, 22, 25, 35, 36, 113
    - filter, 106
    - map, 25, 36, 37, 106
    - reduce, 25, 38
    - sort, 23
  - functional
    - adjoin, 106
    - binaryFun, 14, 83
    - unaryFun, 14, 83
  - range, 16–18, 22, 25, 35, 36, 42, 54
  - traits
    - `isIntegral`, 96
    - `ParameterTypeTuple`, 30, 62
    - `ReturnType`, 23, 30, 37
    - `TypeTuple`, 62
    - `Unqual`, 95
  - typecons
    - `Tuple`, 60–62, 112, 113
    - tuple, 62
  - typetuple, 62, 67, 94
    - `TypeTuple`, 60
- storage class
  - `enum` as a kind of, 10
- string mixins, 58, 72, 76
  - limitations, 79
- struct templates
  - `memoize`, 30, 31
- syntax
  - anonymous function templates, 27
  - class templates, 46
    - parameterized inheritance, 46
  - enum template, 52
  - function templates, 23
  - interface templates, 52
  - `is` expression, 109, 111
  - mixin templates, 69
  - nested `static ifs`, 16
  - `opDispatch`, 71
  - operator overloading, 68
  - static if, 15
  - string mixins, 76
  - struct templates, 34
  - template constraints
    - for class templates, 54
  - template declaration, 8
  - template instantiation, 10, 11
  - templated constructors, 39
  - templates constraints, 53
  - templates specialization, 21
  - union templates, 52
  - Universal Function Call Syntax, 107
- template
  - and CTFE, 80
  - arguments
    - templates as, 11
  - constraints, 53, 57
  - curried templates, 20
  - declaration, 8
  - double-decker template
    - `CheckCompatibility.With`, 33
  - function templates, 23
    - anonymous function templates, 27
  - mixin templates, 69
  - parameters
    - alias, 8, 9, 19, 30, 31, 33, 37, 40, 65
    - as a base class, 48
    - IFTI, 24
    - integral value, 19, 33
    - literal value, 8
    - storage class, 28
    - template this parameter, 42
    - tuple, 8, 59
    - type, 8, 65
    - `void`, 12
  - predicate templates, 15
  - recursion, 17
  - specialization, 20, 53
  - wrapper templates, 73
- templates constraints
  - and `opDispatch`, 71
- templates specialization
  - compared to constraints, 58
- `__traits`, 48, 57
- tuples

- containing `void`, 62
  - empty, 62
  - the many kinds of, 61
- type tuple
  - eliminating occurrence of a type, 66
  - manipulation, 67
  - recursion, 64
- UFCS, 107
- `void`
  - as a template parameter, 12
  - in tuples, 62